

**О. И. Мельников, А. А. Морозов**

**МАТЕМАТИЧЕСКОЕ МОДЕЛИРОВАНИЕ  
С ПРИМЕНЕНИЕМ СИСТЕМЫ MAPLE**

РЕПОЗИТОРИЙ БГУ

УДК 518(075.8)  
ББК 22.191я73  
М801

Рецензенты:

доктор физико-математических наук, профессор Р. И. Тышкевич  
кандидат физ.-мат. наук, доцент Н. В. Костюкевич  
учитель высшей категории М. Н. Мосензон

**Мельников О. И.**

М801 Математическое моделирование с применением системы MAPLE. – Мн.: , 2009.– 100 с.

ISBN 985-435-850-X

Научно-популярным издание по математике и информатике. В доступной форме изложены принципы и приемы построения математических моделей, для реализации и исследования которых применяется система компьютерной алгебры Maple.

Для учащихся 9—11 классов общеобразовательных школ, учителей информатики, студентов естественнонаучных специальностей педагогических вузов.

ISBN 985-435-850-X

**УДК 518(075.8)**  
**ББК 22.191я73**

© О. И. Мельников, А. А. Морозов 2009  
©

### Дорогие друзья!

Книга, которую Вы держите в руках, заинтересует посвящена построению и исследованию математических моделей и их компьютерной реализации. Моделирование заключается в создании некоторых искусственных объектов (моделей), в упрощенном виде описывающих реальные объекты, явления и ситуации. Затем модели изучают, чтобы получить новые сведения о реальности.

Надеемся, что эта книга будет интересна не только любителям математикам, поскольку при построении математических моделей без математики не обойтись, но и тем, кто увлекается информатикой, так как без вычислительной техники исследование математических моделей очень усложняется, а во многих случаях просто невозможно.

В книге приведены первоначальные сведения о моделях и моделировании, указаны основные требования, предъявляемым к математическим моделям. Описаны основные команды и функции системы компьютерной математики Maple, которая используется для реализации и исследования построенных моделей.

Затем показано как с помощью моделирования можно решать экономические, технические и физические задачи, связанные с *определением корней уравнений* или *экстремальных значений функций*. Далее рассмотрены различные математические модели и методы их исследования.

Так, при проведении банковских операций используется *финансовое моделирование*, которое позволяет, определить на каких условиях выгоднее помещать вклады в банк или брать там кредит.

С помощью *биологических* и *экологических моделей* изучают развитие популяций живых организмов в разных условиях, изменение численности и возрастного состава населения, особенности протекания эпидемий, оценивают экологическую обстановку.

На основании *прогностических моделей* исследуют функционирование различных систем и делают выводы, позволяющие предсказать их дальнейшее развитие на протяжении некоторого времени.

*Моделирование случайных процессов* дает возможность выявить некоторые закономерности, используемые для оптимизации управления такими процессами или прогнозирования результатов их протекания.

Обращаем Ваше внимание на то, что весь материал, предложенный в книге, надо проработать, сидя за компьютером. Это будет способствовать приобретению прочных навыков к использованию современных компьютерных технологий для решения конкретных задач, и поможет подготовиться к изучению более сложных математических моделей.

Желаем успеха.

РЕПОЗИТОРИЙ БГПУ

## 1. О МАТЕМАТИЧЕСКОМ МОДЕЛИРОВАНИИ

### 1.1. Что такое математическая модель

Предположим, что Вам поручено разработать новое изделие, обладающее определенными свойствами, или разобраться в какой-то проблемной ситуации, допускающей различные варианты разрешения, понять сущность незнакомого явления или исследовать работу сложной системы и дать рекомендации по ее эффективному управлению. Если данная ситуация Вам незнакома, то не стоит сразу браться за ее разрешение. Надо глубже вникнуть в суть задачи, понять, что в ней главное, а что второстепенное.

Вначале попробуйте выделить основные составляющие изучаемой ситуации и существенные связи между ними, и каким-то образом изобразить все это, возможно в виде копии объекта (системы) или описания ситуации (явления). Если рассматриваемая задача требует выбора наилучшего способа управления ситуацией или системой, то необходимо определить, как можно влиять на ее развитие и что подразумевается под эффективным управлением. Когда все эти этапы будут пройдены, тогда можно сказать, что Вы построили *модель*.

**Модель** — это искусственно созданный образ, который описывает строение и основные свойства реального объекта, системы, ситуации или явления с целью их изучения. Термин «модель» происходит от латинского *modulus* — мера, образец.

Процесс построения моделей называют **моделированием**.

Модели бывают *физическими* (предметными) и *знаковыми* (символьными). Примерами первых являются различные макеты, устройства, образцы, вторых — графики, схемы, программы, математические соотношения.

Так, глобус — это модель Земного шара, карта — модель части земной поверхности. А отправляясь в поход, можно смоделировать будущий путь, т. е. проложить его на карте и заранее подготовиться к возможным препятствиям и трудностям.

Чтобы понять, как будет вести себя самолет в воздушном потоке, изготавливают его уменьшенную копию, которую помещают в аэродинамическую трубу. Наблюдая за поведением модели, можно сделать выводы о движении самолета во время реального

полета. Это и дешевле, и безопаснее, чем испытания несовершенного варианта изделия. Часто архитектор, для того чтобы увидеть, как будет выглядеть проектируемое здание среди уже построенных, создает его копию, которую и размещает на макете застраиваемой территории.

Можно дать еще одно определение модели.

**Модель** — это упрощенный аналог реального объекта (системы, явления, ситуации), передающий его главные особенности и отбрасывающий второстепенные.

Упрощение при создании моделей производится сознательно. Например, тематические географические карты рассказывают о животном и растительном мире, народонаселении и его плотности, климате и т. д. На исторических картах показаны границы уже несуществующих государств или схемы давно произошедших сражений. Каждая из таких моделей части земной поверхности передает те ее особенности, которые важны для составителя карты.

При изготовлении модели самолета учитываются форма самолета и материал, из которого изготовлен его корпус, и не имеют значения, например, внутреннее устройство и оформление салона.

Для архитектора главным в модели являются форма и внешний вид объектов и не важны стоимость, коммуникационные связи и т. п.

## **1.2. Примеры построения некоторых математических моделей**

Среди знаковых выделяют *математические модели*, в которых объекты или ситуации описывают с помощью математических средств.

Встречались ли Вам в жизни математические модели? Конечно! Вспомните любую задачу, связанную с составлением уравнений. Переход от задачи, условие которой задано словами, к математическим уравнениям или неравенствам — простейший пример математического моделирования. При таком моделировании уже изученная ситуация предлагается в виде условия задачи. Решая задачу, Вы сначала определяете неизвестные величины, после составляете уравнения и неравенства, которые связывают неизвестные величины с известными, а затем находите решения полученных уравне-

ний и неравенств, подходящие по смыслу задачи. Первые два этапа решения задачи можно отнести к *построению математической модели*, последний — к *исследованию модели*.

Часто при рассмотрении самых различных вопросов рисуют схемы, в которых людей или населенные пункты обозначают точками, а отношения между людьми или дороги между населенными пунктами — линиями. Эти схемы также являются моделями, ведь на них отмечено главное с точки зрения их составителя.

При решении задач обычно рассматриваются упрощенные ситуации. Например, в задачах на движение считается, что тела мгновенно набирают скорость и меняют направление движения, а скорость течения реки одинакова на протяжении всего пути. В задачах на смеси предполагается, что объем полученной смеси равен сумме объемов ее компонентов, в задачах на производительность, что работы выполняются равномерно. Однако эти допущения не всегда выполняются в действительности.

Таким образом, математические модели различного вида встречаются достаточно часто в функциональных, графических и других выражениях.

При построении математических моделей необходимо выделять основные этапы: изучение реальной ситуации; формализация — переход от ситуации к модели; исследование модели на языке той теории, в рамках которой она построена; интерпретация полученных результатов — истолкование решения в терминах исходной ситуации.

Сегодня моделирование, являясь важным методом изучения реальных явлений, широко используется в науке, технике и экономике. В последнее время математические модели нашли применение в лингвистике, литературоведении, юриспруденции, социологии и других гуманитарных дисциплинах.

Рассмотрим примеры построения и исследования математических моделей.

**Ситуация 1.** Населенные пункты Александровка и Березовка находятся по одну сторону от прямолинейного шоссе. Необходимо найти на шоссе такое место для автобусной остановки, чтобы суммарная стоимость постройки дорог от Александровки до остановки и от Березовки до остановки была наименьшей.




Рис. 1

Смоделируем ситуацию, т. е. заменим населенные пункты точками  $A$  и  $B$ , а шоссе и проектируемые дороги — отрезками прямой. Прямая короче всякой ломаной или кривой, соединяющей концы отрезка, поэтому предполагаем, что стоимость постройки таких дорог будет наименьшей (рис. 1).

На прямой  $p$  необходимо найти такую точку  $C$ , чтобы для любой другой точки  $C_1$  этой прямой выполнялось соотношение:

$$|AC| + |CB| \leq |AC_1| + |C_1B|. \quad (1.1)$$

Для решения задачи построим точку  $B_1$ , симметричную точке  $B$  относительно прямой  $p$  (рис. 2,  $a$ ). Соединим точки  $A$  и  $B_1$  отрезком. Докажем, что точка пересечения этого отрезка и прямой  $p$  является искомой точкой  $C$ .






Рис. 2

Возьмем на прямой  $p$  произвольную точку  $C_1$  (рис. 2, б). Поскольку отрезок короче любой ломаной, соединяющей его концы, то

$$|B_1C| + |CA| < |B_1C_1| + |C_1A|. \quad (1.2)$$

Прямая  $p$  является срединным перпендикуляром к отрезку  $BB_1$ , поэтому

$$|B_1C| = |CB|, |B_1C_1| = |C_1B|.$$

В неравенстве (1.2) заменим  $|B_1C|$  на  $|CB|$  и  $|B_1C_1|$  на  $|C_1B|$ , получим требуемое соотношение (1.1). Утверждение о том, что  $C$  является искомой точкой, доказано.

При построении модели мы заменили населенные пункты точками и не учитывали особенностей местности в районе строительства, хотя неровности рельефа, водные преграды, лесные участки могут сделать неверной нашу гипотезу о том, что стоимость дорог зависит только от их длины.

**Ситуация 2.** Ваш товарищ открыл небольшой мебельный цех и просит помочь ему составить план выпуска продукции, обеспечивающий наибольший доход.

Вначале, естественно, надо определить возможности производителя. Известно, что недельный ресурс составляет 75 погонных метров досок и 276 ч рабочего времени. После этого следует оценить потребности рынка. Оказалось, что наибольшим спросом пользуются столы и шкафы стоимостью 60 и 240 тыс. руб. соответственно. Поэтому было решено выпускать именно эти товары.

Для изготовления стола требуется 2 м досок и 3 ч рабочего времени. Шкаф изготовить сложнее: нужно 5 м досок и 21 ч работы мастеров.

Ваша помощь заключается в определении оптимального количества столов и шкафов, т. е. надо найти ответ на вопрос: сколько следует изготовить столов и сколько шкафов, чтобы полученная от их реализации прибыль оказалась наибольшей.

Предположим, что будет выпущено  $x$  столов и  $y$  шкафов, реализовав которые получим:

$$z = 60x + 240y \text{ тыс. руб.}$$

Нужно так подобрать управляемые переменные  $x$  и  $y$ , чтобы значение функции  $z$  оказалось наибольшим. Очевидно, что при этом величины  $x$  и  $y$  должны быть неотрицательными.

Определим, как имеющиеся в наличии ресурсы влияют на объем выпускаемой продукции. Для выпуска  $x$  столов потребуется  $2x$ , а для выпуска  $y$  шкафов —  $5y$  погонных метров досок. Следовательно, досок необходимо  $2x + 5y$ , и это количество не должно превышать имеющиеся в наличии 75 погонных метров. Аналогично, необходимое для производства количество рабочего времени  $3x + 21y$  не должно быть больше 276.

Следовательно, математическая модель для определения оптимального количества выпускаемой продукции выглядит так: необходимо найти неотрицательные и целочисленные значения  $x$  и  $y$ , при которых значение функции

$$z = 60x + 240y$$

будет наибольшим. При этом должны выполняться следующие условия:

$$\begin{cases} 2x + 5y \leq 75, \\ 3x + 21y \leq 276. \end{cases}$$

Функция, для которой нужно найти наибольшее значение, называется *целевой функцией*.

Обратите внимание, что при построении модели рассматриваются только главные составляющие объектов и их основные свойства, т. е. в данном случае безразлично из какого дерева изготовлены доски или как зовут мастеров, поскольку это не влияет на прибыль. Не учитывается также колебание цен на рынке, затраты на доставку готовой продукции и стоимость отделочных материалов, необходимых для производства столов, возможность того, что станок выйдет из строя из-за поломки и т. д. С одной стороны, некоторые из перечисленных факторов сложно учесть, поскольку они случайны или трудно описываются математически, а с другой стороны, слишком большое число учтенных факторов приводит к усложнению модели и увеличению трудностей ее дальнейшего исследования.

Таким образом, требования к математической модели: *полнота описания* и *простота описания* объекта, явления или ситуации, противоречат друг другу. В процессе построения модели необходимо выбрать золотую середину.

Проведем исследование нашей модели. Поскольку шкафы дороже, то попробуем сначала производить только их. Досок хватит на изготовление  $75 : 5 = 15$  шкафов, однако за рабочее время можно сделать только  $276 : 21 \approx 13$  шкафов. Для столов ресурсов не остается. Проданные шкафы принесут доход  $240 \cdot 13 = 3120$  тыс. руб.

Попробуем повысить доход за счет уменьшения числа шкафов и увеличения числа выпускаемых столов. Изготовим 12 шкафов. Тогда останется  $75 - 12 \cdot 5 = 15$  м досок и  $276 - 12 \cdot 21 = 24$  ч рабочего времени, т. е. мастера могут сделать 8 столов, но досок хватит только на 7. Доход от продажи в этом случае будет  $60 \cdot 7 + 240 \cdot 12 = 3300$  тыс. руб.

Уменьшим число шкафов еще на один. Тогда останется  $75 - 11 \cdot 5 = 20$  м досок и  $276 - 11 \cdot 21 = 45$  ч рабочего времени. За это время можно изготовить 15 столов, но досок хватит только на 10. Полученный доход окажется равным  $60 \cdot 10 + 240 \cdot 11 = 3240$  тыс. руб.

Следовательно, уменьшать число выпускаемых шкафов не рационально. Для получения наибольшего дохода (3300 тыс. руб.) надо выпускать 7 столов и 12 шкафов в неделю.

Существуют более эффективные методы решения подобных задач. Некоторые из них рассмотрены далее во второй части этой книги.

### **Вопросы и задания для самостоятельной работы**

**1.** Что можно считать моделью Каменецкой Белой Вежи: цилиндр, конус, куб, усеченный конус?

**2.** На тестирование по математике пришли ученики из пяти школ: по 20 из первой и пятой, 15 — из второй, 45 — из третьей, 30 — из четвертой. Распределите учеников по 30-местным аудиториям так, чтобы в каждой из них было примерно одинаковое число учеников из всех школ.

**3.** Деревни Ивановичи, Петровичи и Сидоровичи находятся около прямолинейного шоссе. Причем Петровичи, расположенные между двумя другими деревнями, находятся на расстоянии 1 км от Ивановичей и на расстоянии 2 км от Сидоровичей. В Ивановичах живут 50 школьников, в Петровичах — 100, в Сидоровичах — 200. В каком месте нужно построить школу, чтобы сумма длин путей всех учеников до школы была наименьшей? А если в каждой деревне живет по 100 школьников? А если в Ивановичах и Петровичах по 100 учеников, а в Сидоровичах — 200?

**4.** Прямолинейный канал, ширина которого 4 м, поворачивает под прямым углом. Есть две прочных доски длиной по 3,5 м. Как с их помощью перейти канал?

## 2. ЗНАКОМСТВО С MAPLE

### 2.1. Основные команды и функции

**Maple** — система компьютерной математики, разработанная канадской фирмой Waterloo Maple, которая ориентирована на широкий круг пользователей: от школьников до преподавателей вузов и научных работников. Она известна благодаря своей высокой эффективности при решении самых разных математических задач и великолепной графике. Свыше 300 крупнейших университетов мира пользуются этой системой. Познакомимся с основными командами и функциями Maple, которые в этой книге используются для реализации и исследования математических моделей.

#### Начало работы

После запуска Maple загружается новый *рабочий лист* (*worksheet*), на котором можно размещать математические выражения (формулы), команды, абзацы с текстом, написанным различными стилями. Строки, начинающиеся с приглашения — знак больше, обозначают клавиатурный ввод. Они обычно выделены красным цветом. Мы задаем формулу или команду, нажимаем клавишу **<Enter>** и получаем результат (числовое значение, выражение, график) либо сообщение об ошибочно набранной команде. Ответы системы выделены синим цветом. Сразу же в новой строке выдается приглашение вводить следующую команду и т. д. Вместо **<Enter>** можно левой клавишей мыши нажать кнопку с восклицательным знаком, которая находится на *панели инструментов* в верхней части окна Maple.

Рассмотрим несколько простых примеров:

```

[ > 2 + 2;
  ]
[ > 2 + 2 + a;
  ]
[ >
  ]

```

4  
4 + a

*Ввод* и соответствующий ему *вывод* на рабочем листе отмечаются квадратной скобкой слева. Ее длина зависит от размера выражений. Обратите внимание на завершающий символ команды. Если в конце поставить точку с запятой, то после выполнения команды выводится ответ. Если набрать только  $2 + 2$  без каких-либо дополнительных сим-

волов, то Maple выдаст предупреждающее сообщение «Warning, premature end of input» («Предупреждение, преждевременное окончание ввода»).

В командную строку можно записать любое выражение. В одной строке для ввода можно располагать несколько команд, разделенных точкой с запятой или двоеточием. Двоеточие в конце команды указывает на то, что ответ на экран не выводится. Это позволяет избежать чрезмерного заполнения рабочего листа результатами промежуточных вычислений.

Чтобы выполнить все вычисления заново в том порядке, в котором они записаны на рабочем листе, нужно на панели инструментов нажать левой клавишей мыши кнопку с тремя восклицательными знаками. Вместо этого можно воспользоваться пунктом меню **Edit>Execute>Worksheet** (Правка> Выполнить> Рабочий лист).

### Арифметические операции и операции отношения

При клавиатурном вводе основных арифметических действий и операций отношения используются следующие символы: + (сложение), - (вычитание), \* (умножение), / (деление), ^ (возведение в степень), ! (факториал), < (меньше), <= (меньше или равно), > (больше), >= (больше или равно), = (равно), <> (не равно).

### Буквы греческого алфавита

Кроме букв латинского алфавита в Maple можно выводить буквы греческого алфавита. Для этого достаточно в командной строке набрать название соответствующей буквы. Например, буква  $\alpha$  получится, если набрать alpha.

Названия греческих букв:

$\alpha$  — alpha,  $\beta$  — beta,  $\gamma$  — gamma,  $\delta$  — delta,  $\epsilon$  — epsilon,  $\zeta$  — zeta,  $\eta$  — eta,  $\theta$  — theta,  $\iota$  — ita,  $\kappa$  — kappa,  $\lambda$  — lambda,  $\nu$  — nu,  $\mu$  — mu,  $\xi$  — xi,  $\pi$  — pi,  $\rho$  — rho,  $\sigma$  — sigma,  $\upsilon$  — upsilon,  $\phi$  — phi,  $\chi$  — chi,  $\psi$  — psi,  $\omega$  — omega.

Чтобы получить заглавную (прописную) греческую букву, например  $\Omega$ , следует набрать Omega. В последних версиях Maple греческие буквы также можно набирать с помощью специального меню.

### Некоторые математические функции

Экспонента и логарифм:

- $\exp(x)$  — экспонента (функция  $e^x$ );

- $\ln(x)$  — натуральный логарифм  $x$ .

Тригонометрические функции:

- $\cos(x)$  — косинус  $x$ ;
- $\sin(x)$  — синус  $x$ ;
- $\tan(x)$  — тангенс  $x$ .

Прочие функции:

- $\text{abs}(x)$  — абсолютная величина (модуль)  $x$ ;
- $\text{sqrt}(x)$  — квадратный корень из  $x$ ;
- $\text{surd}(x, n)$  — корень степени  $n$  из  $x$ .

Таким образом, с программой Maple можно работать как с калькулятором.

### Функция `evalf`

Все вычисления в Maple производятся в аналитическом (символьном) виде. Это означает, что точный результат будет содержать иррациональные выражения, а также константы, такие как  $\pi$ ,  $e$  и др. Чтобы получить приближенный числовой результат в привычном виде, следует использовать функцию `evalf`. Например:

```
> sqrt(2); evalf(sqrt(2), 5);
```

$$\sqrt{2}$$

$$1.4142$$

Второй параметр функции `evalf` указывает на количество цифр приближенного значения, начиная с первой ненулевой цифры слева (т. е. это не число цифр после десятичной точки). Если параметр не задан, то предполагается стандартное значение количества цифр, равное 10.

### Рациональные и вещественные числа

Рациональные числа в Maple представлены в виде дробей с использованием операции деления, вещественные числа — в форме с десятичной точкой. Например:

```
> 10/3, evalf(10/3, 5), 1.2*10^-2;
```

$$\frac{10}{3}, 3.3333, 0.0120000000$$

Есть и другой способ получить ответ в виде десятичной дроби. Найдем, например, корень квадратный из 10:

```
[ > sqrt(10.0);
                                     3.162277660
```

То есть, если в выражение входит вещественное число, то и значение этого выражения дается в том же виде.

### Функции округления и отсечения

Функции округления и отсечения нужны в вычислениях, где используются вещественные числа:

- `ceil(x)` — наименьшее целое, большее или равное  $x$ ;
- `floor(x)` — наибольшее целое, меньшее или равное  $x$ ;
- `round(x)` — округление  $x$  по правилам математики;
- `frac(x)` — дробная часть  $x$ ;
- `trunc(x)` — отбрасывание дробной части  $x$ .

Примеры:

```
[ > ceil(1.5), floor(1.5), round(1.49), round(1.5);
                                     2, 1, 1, 2
[ > frac(1.5), trunc(1.5);
                                     0.5, 1
```

### Большие и малые буквы при наборе команд

Из-за невнимательности при наборе команд иногда можно получить непонятные результаты и трудно находимые ошибки. Следует помнить, что большие и малые буквы Maple воспринимает по-разному. Например, **Pi** и **pi** дадут на экране одинаковое изображение  $\pi$ , хотя в первом случае это будет известная константа, а во втором — новая переменная. Проверить это можно следующим образом:



```

[ > Pi; evalf(Pi);
                                     π
                                     3.141592654
[ > pi; evalf(pi);
                                     π
                                     π

```

Вычислим сумму квадратов целых чисел от 1 до 10:

```

[ > sum(k2, k = 1 .. 10);
                                     385

```

Здесь две точки, идущие подряд, определяют диапазон значений целочисленной переменной  $k$ . Если же команду суммирования записать большими буквами, то в ответе Maple выдаст исходное выражение:

```

[ > SUM(k2, k = 1 .. 10);
                                     SUM(k2, k = 1 .. 10)

```

Это говорит о том, что такая команда системе неизвестна.

### Оператор присваивания

Оператор присваивания  $:=$  набираем на клавиатуре при помощи знаков двоеточие и равно. Применяя присваивание, можно заменить любое выражение переменной. Например:

```

[ > S := Pi · R2 : S;
                                     π R2

```

Дадим  $R$  конкретное значение. Текущее символьное значение переменной  $S$  также заменится на новое значение:

```

[ > R := 2 : S;
                                     4 π
[ > evalf(%, 5);
                                     12.566

```

Последняя команда в этом примере выводит приближенное значение  $S$ . Символ процента служит в Maple для обозначения результата выполнения предыдущей команды (часто это позволяет упростить запись выражений и отказаться от ненужных вспомогательных переменных).

Если при наборе команды одновременно нажать клавиши **<Shift>** и **<Enter>**, то курсор переходит на следующую строку, не обрывая при этом ввода. Это удобно, когда надо расположить несколько команд одна под другой, а также при выравнивании текста длинных команд, не помещающихся в одной строке. Пример вычисления суммы ряда:

```
[ > s := sum(  $\frac{x^k}{k!}$ , k = 0 .. 5 );
  evalf(subs(x = 1, s));
                                      $s := 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5$ 
                                     2.716666667
```

Здесь команда **subs** используется для подстановки в выражение  $s$  значения  $x = 1$ .

### Числа с произвольным количеством цифр

Вычислим число  $\pi$  с 50 значащими цифрами:

```
[ > evalf(Pi, 50);
                                     3.1415926535897932384626433832795028841971693993751
```

Вычислим 50!:

```
[ > 50!;
                                     30414093201713378043612608166064768844377641568960512000000000000
```

и разложим полученное число на простые множители:

```
[ > ifactor(%);
                                      $(2)^{47} (3)^{22} (5)^{12} (7)^8 (11)^4 (13)^3 (17)^2 (19)^2 (23)^2 (29) (31) (37) (41) (43) (47)$ 
```

### Алгебраические преобразования

Рассмотрим, как система Maple справляется с алгебраическими преобразованиями выражений.

Для раскрытия скобок предназначена команда **expand**, которая может применяться к выражениям любых типов. Примеры:

```
[ > expand(sin(x + y));
                                      $\sin(x) \cos(y) + \cos(x) \sin(y)$ 
```

```
[> expand((a-b)^3);
```

$$a^3 - 3a^2b + 3ab^2 - b^3$$

Разложим на множители полученный результат:

```
[> factor(%);
```

$$(a - b)^3$$

Напоминаем, что символ % означает подстановку результата выполнения предыдущей команды.

Команда **factor** используется также для приведения к общему знаменателю суммы рациональных выражений. Например:

```
[> factor(1/a + 1/b);
```

$$\frac{b+a}{ab}$$

Для сокращения алгебраической дроби может применяться команда **normal**:

```
[> normal((a-b)/(a^2-b^2));
```

$$\frac{1}{b+a}$$

Для приведения подобных членов используется команда **collect**:

```
[> expr := ln(x) - 3*a^2*ln(x);
collect(expr, ln(x));
```

$$(1 - 3a^2) \ln(x)$$

С помощью команды **simplify** Maple позволяет успешно справляться с упрощением длинных (например, тригонометрических) выражений. Пример:

```
[> simplify(cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos(2*x));
```

$$\cos(x)^4 (\cos(x) + 1)$$

### Наименьшее и наибольшее значения

Определение наименьшего значения. Пример:

```
[> min(5, 2, 3);
```

$$2$$

Какое число больше,  $\pi^e$  или  $e^\pi$ ?

```
[ > max(Pi^exp(1), exp(1)^Pi);
                                     (e)π
```

Обратите внимание, что для вычисления  $e^x$  надо записывать не  $e^x$ , а  $\text{exp}(x)$ .

### Команда restart

После набора и выполнения некоторой команды или группы команд всегда можно вернуться на ранее введенную строку рабочего листа, и после нажатия на <Enter> повторить вычисления. При этом нередко результат будет отличаться от предыдущего, так как одни и те же обозначения в разных частях документа Maple могут иметь разный смысл.

Поэтому при решении каждого нового примера рекомендуется, прежде всего, выполнить команду **restart**, которая уничтожит значения всех переменных, используемых в открытых документах текущего сеанса работы. То есть начать вычисления «с чистого листа».

Удалить значение отдельной переменной можно также с помощью такого присваивания:

*переменная := 'переменная'*

Например:

```
[ > x := 1: x + 1; x := 'x': x + 1;
                                     2
                                     x + 1
```

Далее  $x$  следует рассматривать как переменную, значение которой не определено.

### Способы задания функций

В системе Maple функции обычно задаются в виде подстановки:

$f := (x, y, \dots) \rightarrow \text{выражение}$

Стрелку набираем на клавиатуре при помощи знаков минус и больше. После этого вызов функции  $f$  записывается, в традиционном виде  $f(a, b, \dots)$ . Переменные  $x, y, \dots$ , указанные в списке параметров, являются *локальными*. Это означает, что за пределами выражения они либо не определены, либо сохраняют ранее присвоенные им значения. Примеры:

```

[ > f := x → sqrt(1 + x^2);
                                     f := x → √(1 + x^2)
[ > f(1); f(1.0);
                                     √2
                                     1.414213562
[ > g := (x, y) → sqrt(x^2 + y^2);
    g(1, a);
                                     √(1 + a^2)

```

Математическое выражение можно преобразовать в функцию с помощью команды **unapply**, например:

```

[ > f := unapply(x^2 + x - 2, x);
                                     f := x → x^2 + x - 2

```

### Решение уравнений, неравенств и их систем

В Maple универсальным средством, позволяющим решать в аналитическом виде уравнения, неравенства и их системы, является команда **solve**. Для записи отношения равенства в уравнениях используется знак равно. Результат выдается в виде последовательности выражений. Найдем, например, корни уравнения  $x^2 - a = 0$  с неизвестным  $x$  и параметром  $a$  (можно также рассматривать  $a$  как неизвестное, а  $x$  как параметр):

```

[ > sol := solve(x^2 - a = 0, x);
                                     sol := √a, -√a

```

Если для дальнейших вычислений необходимо иметь значение одного из корней в виде отдельной переменной, то его всегда можно извлечь с помощью индекса:

```

[ > x1 := sol[1]; x2 := sol[2];
    x1, x2;
                                     √a, -√a

```

Для проверки подставим значение корня в исходное уравнение:

```

[ > subs(x = sol[1], x^2 - a = 0);
                                     0 = 0

```

Полученное равенство говорит о правильности ответа.

Maple может находить как вещественные, так и комплексные корни алгебраических уравнений. Например:

```
[ > solve(x3 + x - 2 = 0, x);
      1, -1/2 + 1/2 I√7, -1/2 - 1/2 I√7 ]
```

В ответе содержится мнимая единица, изображаемая I.

Решим с помощью команды **solve** систему двух уравнений:

```
[ > sols := solve({x + 2·y = 3, 1/x + y = 1}, {x, y});
      sols := {x = -1, y = 2}, {x = 2, y = 1/2} ]
```

В ответе два набора решений. Можно выделить каждый из них:

```
[ > sols[1]; sols[2];
      {x = -1, y = 2}
      {x = 2, y = 1/2} ]
```

Далее присвоим полученные значения переменным x1, y1, x2, y2. Это делается с помощью команды **subs**:

```
[ > x1 := subs(sols[1], x) : x2 := subs(sols[2], x) :
      y1 := subs(sols[1], y) : y2 := subs(sols[2], y) :
      x1, x2, y1, y2;
      -1, 2, 2, 1/2 ]
```

Следующие примеры показывают, как легко решать в Maple неравенства и системы неравенств:

```
[ > solve({x2 - 1 ≤ 0}, x);
      {-1 ≤ x, x ≤ 1}
[ > solve({x2 < 1, y2 < 1, x + y < 1/2}, {x, y});
      {-1 < x, -1 < y, x < 1, y < 1, x + y < 1/2} ]
```

### Численное решение уравнений

Когда Maple не может найти ни одного аналитического (в виде формулы) решения уравнения, то можно попробовать найти корни (если они существуют) в численном виде с помощью команды **fsolve**. В общем случае она вычисляет один корень, но для многочленов определяет все действительные корни. Например:



```

[> seq(i, i = 1..10, 2);
                                     1, 3, 5, 7, 9
[> seq(-0.5..0.5, 0.1);
                                     -0.5, -0.4, -0.3, -0.2, -0.1, 0., 0.1, 0.2, 0.3, 0.4, 0.5
[> seq(i^2, i = 1..5);
                                     1, 4, 9, 16, 25
[> seq(1/k, k = 1..10);
                                     1, 1/2, 1/3, 1/4, 1/5, 1/6, 1/7, 1/8, 1/9, 1/10
[> seq(sin(Pi/6*i), i = 0..6);
                                     0, 1/2, 1/2*sqrt(3), 1, 1/2*sqrt(3), 1/2, 0

```

Чтобы получить последовательность пронумерованных переменных или выражений, необходимо использовать операцию сцепления, которая обозначается двумя вертикальными черта, например:

```

[> seq(x||i, i = 1..10);
                                     x1, x2, x3, x4, x5, x6, x7, x8, x9, x10
[> x2 := 1 : x2;
                                     1

```

### Списки и множества

В Maple важная роль отводится *спискам* и *множествам* — структурам данных, получаемым из последовательности. Последовательность выражений, заключенная в квадратные скобки, образует список, последовательность, заключенная в фигурные скобки, — множество. Списки и множества используются для хранения произвольных наборов данных. Рассмотрим примеры.

Пустой список и пустое множество:

```

[> lst1 := [] : set1 := {} :

```

При задании списков важен порядок элементов. Отдельные элементы списка выделяются с помощью индекса. Им можно присваивать новые значения, например:



```
[> lst1 := [1, 1 + x, 0, sqrt(x)]: lst1[1] := x :
  lst1;
      [x, 1 + x, 0, sqrt(x)]
```

При выборе нескольких элементов списка можно использовать диапазоны индексов:

```
[> lst1[2..4];
      [1 + x, 0, sqrt(x)]
```

В списке все повторяющиеся значения сохраняются, тогда как во множестве им будет соответствовать один элемент, поэтому при задании множеств порядок элементов не важен, например:

```
[> seq1 := 1, 2, 1, 2, 1, 2 : lst1 := [seq1] : set1 := {seq1} :
  lst1; set1;
      [1, 2, 1, 2, 1, 2]
      {1, 2}
```

### Команда map

Часто возникает необходимость выполнить какую-либо команду или вычислить функцию применительно к каждому элементу списка или множества. В Maple имеется несколько соответствующих команд, одна из них — **map**:

```
[> map(x → x2, [seq(0..5)]);
      [0, 1, 4, 9, 16, 25]
[> map(sin, {x, y, z});
      {sin(x), sin(y), sin(z)}
```

### Массивы

Более сложная структура данных по сравнению со списком — *массив* (array). Индексы массива могут принимать положительные, отрицательные и нулевые значения. Используя вычисляемые индексы, можно с помощью операции присваивания заменить значение любого элемента массива. Массив также может иметь несколько размерностей, каждая со своим индексом.

Для задания массива служит команда **array**, например:

```

> a := array(0..3);
  a[0] := sqrt(x);
                                     a := array(0..3, [ ])
                                     a0 := √x

> a := array(-1..1, [seq(2..4)]);
  a[0];
                                     3

> T := array(1..3, 1..3);
  T[1, 2] := 2;
                                     T := array(1..3, 1..3, [ ])
                                     T1,2 := 2

```

Массивы удобны для хранения и обработки больших наборов данных. Очень часто они используются при построении различных таблиц. Для вывода значений массива используется командой **print**, например:

```

> v := array(1..3, [1, 2, 3]); v;
                                     v
> print(v);
                                     [ 1 2 3 ]

```

Элементом массива может быть список, элементом списка — массив и т. д. Комбинирование этих возможностей позволяет создавать довольно сложные структуры данных, ограниченные лишь объемом памяти компьютера.

## 2.2. Программирование в Maple

К программированию прибегают в тех случаях, когда встроенных средств Maple не хватает для решения поставленных задач. Иногда трудно или вообще невозможно обойтись без программирования, например, при выполнении громоздких повторяющихся вычислений.

Система Maple имеет свой интуитивно понятный язык программирования, на котором можно записать как численные расчеты, так и символьные преобразования. Его основные конструкции аналогичны соответствующим средствам традиционных языков программирования. Кратко рассмотрим те из них, которые используются в этой книге.

## Конструкция условного выбора

Как и любой язык программирования, Maple содержит конструкцию *условного выбора* **if**. Приведем примеры:

```
> a := 5 :
   if a < 5 then (a + x)2 else (a + x)3 end if;
                                     (5 + x)3

> a := 2 : b := 3 :
   if a > b then
     print("a больше b");
   else
     if a < b then
       print("a меньше b");
     else
       print("a равно b");
     end if;
   end if;
                                     "a меньше b"
```

Сначала вычисляется *условие* — логическое выражение, записанное после служебного слова **if**. Если условие подтверждается, то выполняются команды, указанные после служебного слова **then**, в противном случае выполняются команды, указанные после слова **else** (эта последняя часть конструкции выбора необязательна).

## Конструкции повторения

К конструкциям повторения относятся различные виды *циклов*. Пример цикла *с условием на продолжение* (цикла **while**):

```
> a := 4 :
   while a > 0 do
     print(a);
     a := a - 1;
   end do;
                                     4
                                     3
                                     2
                                     1
```

Сначала вычисляется *условие* — логическое выражение, записанное после служебного слова **while**. Если условие подтверждается, то выполняются указанные в цикле ко-

манды и все повторяется, в противном случае цикл считается завершенным. Так как все начинается с проверки условия, то команды, записанные в цикле, могут ни разу не выполниться.

Для задания *цикла с параметром* (цикла **for**) необходимо указать *верхнюю и нижнюю границы*, а также величину *шага* для *управляющей переменной* (параметра), например:

```
> for i from 0 by 2 to 5 do
    print(i);
end do:
0
2
4

> for i from 10 by -2 to 6 do print(i) end do:
10
8
6
```

Сначала вычисляются выражения, которые определяют начальное и конечное значения управляющей переменной (после служебных слов **from** и **to**), а также шаг ее изменения (после **by**). Каждое следующее значение этой переменной получается сложением ее предыдущего значения и шага. Затем выполняются указанные в цикле команды. Если начальное значение управляющей переменной больше конечного, а шаг положительный, то такой цикл не выполняется ни разу. Если начальное значение меньше конечного, а шаг отрицательный, то такой цикл не выполняется ни разу. Величину шага можно не задавать, в этом случае предполагается, что она равна единице.

Пример программы, которая формирует цепную дробь:

```
> s := x:
for i from 1 to 4 do s := 1 + 1/s end do:
s;
```

$$1 + \frac{1}{1 + \frac{1}{1 + \frac{1}{1 + \frac{1}{x}}}}$$

### Команда `break`

Команда **break** прекращает выполнение любого цикла. Например, с помощью следующей программы:

```
> for k from 106 to 2·106 do
    if isprime(k) then
        prim := k; break;
    end if;
end do:
prim;
                                     1000003
```

находим первое простое число, которое больше 1 000 000, но меньше 2 000 000.

Здесь стандартная логическая функция **isprime** ( $k$ ) дает значение **true** (истина), если  $k$  — простое число, и **false** (ложь), если  $k$  — число составное.

### 2.3. Построение графиков

Система Maple известна своими богатыми графическими возможностями. В ее библиотеках имеется большое число графических команды самого различного назначения: от рисования двухмерных кривых до изображения сложных трехмерных поверхностей и анимации. Рассмотрим на простых примерах построение наиболее часто встречающихся типов графиков.

#### Команда `plot`

С помощью команды **plot**, которая содержится в основной библиотеке команд Maple, можно начертить график одной или нескольких функций, заданных в декартовых и других координатах, а также отобразить множество отдельных точек. Наиболее часто эта команда записывается в таком виде:

**plot**( $f(x)$ ,  $x = a \dots b$ ,  $c \dots d$ , *дополнительные параметры*).


Здесь  $f$  — функция независимой переменной  $x$ , график которой необходимо отобразить;  $a \dots b$  — диапазон изменения  $x$  по горизонтальной оси графика;  $c \dots d$  — диапазон изменения значений функции по вертикальной оси.

Параметр  $c \dots d$  необязателен, как и *дополнительные параметры*, или *опции*, задаваемые в виде равенств *параметр = значение*.


Дополнительные параметры определяют внешний вид графика: цвет, тип и толщину линий, рамку, разметку осей, координатную сетку, поясняющие надписи и многое другое. После того как график построен, можно изменить его настройки с помощью основ-

ного меню системы Maple или кнопок ее панели инструментов. Набор возможных параметров во всех командах двумерной графики, за некоторым исключением, одинаков. Соответствующие им значения можно найти с помощью справочной системы.

Вот как можно получить график функции  $\sin x$  с делениями на координатных осях:




Если стандартный вид графика Вас не устраивать, то исходя из собственных предпочтений и особенностей решаемой задачи, необходимо определить значения дополнительных параметров. Например, тот же график можно отобразить и в таком виде:



Здесь параметр **color** задает цвет синусоиды, параметр **thickness** — толщину линии, **ytickmarks** — число координатных отметок на вертикальной оси. Параметры **title** и **labels** нужны для нанесения надписей заголовка и осей. Параметры **titlefont** и **labelfont** определяют используемый шрифт, его жирность (**bold**) и размер (**12** пт).

### Разметка координатных осей


Для нанесения разметки на координатные оси используется параметр **tickmarks**, Например:



На этом графике отметки делений по оси  $x$  задаются с промежутком (параметр **spacing**), равным  $\frac{1}{2}\pi$ , числовые отметки по оси  $y$  — списком значений с шагом 0,5.

### Несколько кривых на графике

Довольно часто необходимо отобразить на одном рисунке графики нескольких зависимостей. Команда **plot** позволяет совместить графики функций, заданных списком:




На полученном рисунке видим три синусоиды. Чтобы графики отличались один от другого, они могут быть выделены цветом (параметр **color**) и изображены разными типами линий (**linestyle**).

### Кривые, задаваемые параметрическими уравнениями

С помощью команды **plot** можно построить график функции, задаваемой параметрическими уравнениями:  $x = x(t)$ ,  $y = y(t)$ , где  $t$  — вспомогательная переменная (параметр).

Такие кривые могут быть замкнутыми и самопересекающимися. Например, вычерчим одну из так называемых *фигур Лиссажу*, задаваемую уравнениями:  $x(t) = \sin 2t$ ,  $y(t) = \cos 3t$ .

```
> plot([sin(2*t), cos(3*t), t=-Pi..Pi],
       color=blue, thickness=2, axes=boxed,
       xtickmarks=5, ytickmarks=5,
       axis=[gridlines=[15, color=green, linestyle=3]]);
```



Здесь параметр **axes** со значением **boxed** отвечает за прорисовку рамки, в которую помещается график. Параметр **axis** нужен для того, чтобы отобразить штрихами зеленого цвета линии координатной сетки (**gridlines**).

Точно так же по уравнениям  $x(t) = r \cos t$ ,  $y(t) = r \sin t$  можно вычертить окружность радиусом  $r$ .

### Графики, построенные по заданным точкам


С помощью команды **plot** можно соединить отрезками прямой точки, координаты которых заданы списком  $[[x_1, y_1], [x_2, y_2], \dots]$ . Пусть, например, имеется набор точек:

```
[> data := [-2, 3], [-1, 4], [1, 3], [1, 2], [1, 1], [2, 1], [3, 4]:
```

На графике получим ломаную линию:




```
> plot([data], x=-3..4, 0..5, color=blue, thickness=2,
      tickmarks=[[seq(-3..4)], [seq(0..5)]],
      axis=[gridlines=[15, color=green, linestyle=3]]);
```



Чтобы отобразить на рисунке только точки, необходимо дополнительно указать параметр **style** со значением **point**:

```
> plot([data], x=-3..4, 0..5, color=blue,
      style=point, symbol=circle, symbolsize=18,
      tickmarks=[[seq(-3..4)], [seq(0..5)]],
      axis=[gridlines=[15, color=green, linestyle=3]]);
```



Точки отмечаются символом, который задается параметром **symbol**. Возможные значения этого параметра: **box** — прямоугольник, **cross** — крест, **circle** — окружность, **point** — точка, **diamond** — ромб. Параметр **symbolsize** определяет размер символа.

### Другие графические команды

Команда **plot** позволяет строить графики наиболее распространенных типов. Однако для формирования и вычерчивания графиков специальных видов могут понадобиться команды, которые содержатся в дополнительных библиотеках Maple, называемых

пакетами. Пакеты необходимо загружать с помощью команды **with**, например:

```
[> with(plots) :
```

```
[> with(plottools) :
```

Пакет **plots** включает команды, позволяющие строить двух- и трехмерные графики при решении специальных прикладных задач.


Команды пакета **plottools** предназначены для создания простых геометрических фигур, из которых можно составить более сложные. К ним относятся: точки, отрезки прямой, прямоугольники, многоугольники, круги, эллипсы, дуги, секторы, стрелки, текст и др. Далее рассмотрим графические команды, используемые в этой книге.

### Ломаная линия

Для вычерчивания ломаной линии можно воспользоваться командой **listplot** из пакета **plots**. Координаты вершин ломаной линии на оси  $y$  должны быть заданы списком вида  $[y_1, y_2, \dots]$ . Координаты вершин на оси  $x$  — это порядковые номера соответствующих значений из списка. Пусть, например, имеется набор чисел:

```
[> data := 1, 3, 3, 2, 5, 1, 2, 5, 3, 1 :
```


После подключения пакета **plots**:



на рисунке получим ломаную линию.

## Кривые, задаваемые уравнениями

Команда **implicitplot** из пакета **plots** позволяет строить графики кривых, задаваемых уравнениями, т. е., не выражая зависимость одной переменной от другой в явном виде. Приведем пример. Команда, которая вычерчивает окружность, определяемую уравнением  $x^2 + y^2 = 1$ :




Получим окружность радиусом 1 с центром в начале координат.

## Геометрические фигуры

Для создания простейших геометрических фигур необходимо воспользоваться командами пакета **plottools**. Рассмотрим несколько примеров.

Чтобы разместить на рисунке несколько фигур (а также несколько графиков, полученных при помощи других команд), здесь и далее используется команда **display** из пакета **plots**. Она обеспечивает вычерчивание всех сформированных графиков. Нарисуем два прямолинейных отрезка (**line**), каждый из которых соединяет две точки с заданными координатами:


```
> with(plots): with(plottools):
gr1 := line([0, 0], [2, 2], color=blue, linestyle=1):
gr2 := line([0, 2], [2, 0], color=red, linestyle=3):
display([gr1, gr2], view=[-0.2..2.2, -0.2..2.2], thickness=2);
```



Параметр **view** команды **display** задает границы отображения графиков по осям  $x$  и  $y$ .

Еще один пример. Нарисуем две окружности (**circle**), задаваемые координатами их центров и радиусами, центры окружностей отмечены отдельными точками (**point**):


```
> with(plots): with(plottools):
gr1 := circle([1, 1], 1, color=blue, linestyle=1):
gr2 := circle([0.5, 1], 0.5, color=red, linestyle=3):
gr3 := point([1, 1], color=blue, symbol=circle, symbolsize=20):
gr4 := point([0.5, 1], color=red, symbol=circle, symbolsize=20):
seq1 := seq(gr||i, i=1..4):
display([seq1], view=[-0.2..2.2, -0.2..2.2], thickness=2);
```



Наряду с командой **plot** для вычерчивания ломаной линии может использоваться и команда **curve**. Координаты вершин ломаной должны быть заданы списком вида  $[[x_1, y_1], [x_2, y_2], \dots]$ . Пусть, например, имеется набор точек:

```
> data := [-1, -1], [-1, 1], [1, 1], [1, -1], [-1, -1]:
```

```
> with(plots): with(plottools):
  gr1 := curve([data], color=blue, thickness=2):
  display(gr1, view=[-1.2..1.2, -1.2..1.2]);
```




На рисунке отдельные звенья ломаной образуют квадрат.

### Отображение текста на графиках

Графики, как правило, снабжают различного рода надписями и текстовыми пояснениями. Чтобы вывести текст необходимо воспользоваться командой **textplot** из пакета **plots**, например:

```
> with(plots):
  gr1:= plot(sin(x), x=0..2*Pi, color=blue, thickness=2):
  gr2:= plot([Pi/2, 1], [2.5, 1.4], style=line, color=black):
  gr3:= textplot([2.5, 1.4, "Точка максимума"],
    color=blue, align=right, font=[COURIER, 13]):
  display([seq(gr[i], i=1..3)], view=[-0.2..6.5, -1..1.6]);
```




Положение текста определяется координатами *опорной точки*. Параметр **align** указывает на смещение текста вправо (**right**) или влево (**left**) относительно этой точки.

Текст также можно разместить выше (**above**) или ниже (**below**) опорной точки. Если параметр **align** не задан, то текст выравнивается относительно центра опорной точки.

### Размещение графиков в виде таблицы

При использовании команды **display** несколько графиков можно разместить в ряд или в виде таблицы. Для этого необходимо сформировать одномерный или двумерный массив, содержащий графики, например,  $y = \sin x$ ,  $y = \sin 2x$ :

```
> with(plots) :
   gr1:= plot(sin(x), x=-Pi..Pi, color=blue) :
   gr2:= plot(sin(2*x), x=-Pi..Pi, color=blue) :
   grarray:=array(1..2, [gr1, gr2]) :
   display(grarray, view=[-Pi..Pi, -1..1], thickness=2) ;
```



В данном случае на рисунке две синусоиды выстроены в одну линию.


### Более сложные примеры

**Муаровый узор.** Этот эффект возникает на экране при отображении графиков прямых или кривых линий, когда они пересекаются под небольшими углами. В простейшем случае для получения муарового узора можно начертить отрезки, расходящиеся из одной точки.

Сначала формируем последовательность, содержащую набор отрезков (**line**). Все они имеют общую точку в начале координат:

```
> with(plots) : with(plottools) :
   seq1:= seq(line([0,0], [k,1], color=black), k=-20..20) :
```

Вычерчиваем линии:



На рисунке видим муаровый узор.

**Штрихование области.** При вычерчивании графиков для наглядности часто требуется заштриховать области, получаемые на рисунке. В этом примере строится вертикальная штриховка области, ограниченной волной синусоиды на отрезке от 0 до  $2\pi$ .

Точки, через которые проходят линии штриховки, могут иметь координаты:


$$\frac{2\pi}{n}k, k = 1, \dots, n-1,$$

где  $n$  — количество точек ( $n$  лучше подобрать экспериментально).

Формируем последовательность, которая содержит линии штриховки:

```
> with(plottools) : with(plots) :
  n:= 80 : f:= k->2*Pi*k/n:
  lines:= seq(line([f(i), 0], [f(i), sin(f(i))]), i=1..n-1) :
```

Вычерчиваем на графике синусоиду и штриховку:



**Гистограмма числовой последовательности.** Сформируем последовательность чисел по следующим правилам. Пусть  $n$  — целое положительное число. Если  $n$  — четное, заменим его на  $\frac{n}{2}$ , если  $n$  — нечетное, заменим его на  $3n + 1$ . Повторим то же самое для вновь полученного числа и т. д. Например, начиная с  $n = 10$ , последовательно получим: 5, 16, 8, 4, 2, 1. Если выбрать другое  $n$ , то последним числом в последовательности снова будет 1. Не найдено ни одного начального числа, которое не приводило бы к 1.

Последовательность чисел можно графически изобразить в виде *гистограммы* — ряда столбиков с общим основанием и высотой, соответствующей числовым значениям. Пусть  $k$  — порядковый номер числа  $m$  в последовательности. На координатной плоскости число  $m$  будем представлять вертикальным отрезком, координаты концов которого  $(k, 0)$  и  $(k, m)$ .

Вычислим, например, последовательность для  $n = 250$ :

```

> n:= 250: k:= 1:
  Последовательность:= n:
  while n>1 do
    k:= k+1:
    if n mod 2 = 0 then n:= n/2 else n:= 3*n+1 end if:
    Последовательность:= Последовательность,n
  end do:

> k, max(Последовательность);

```

110, 9232




Следовательно, при  $n = 250$  для достижения 1 необходимо сделать 110 шагов, а максимальное число в последовательности равно 9232.

Теперь сформируем набор отрезков:

```
> with(plots) : with(plottools) :
   seq1:=seq(line([i,0],[i,Последовательность[i]]),i=1..k) :
```

и вычертим гистограмму:

```
> display([seq1],view=[0..k+1,-500..9500],
          color=blue,thickness=2,
          axis[2]=[gridlines=[10,color=green,linestyle=3]]) ;
```



Из рисунка видно, что наибольшее число в последовательности получается после 75-го шага. Найти это число можно, вычислив значение элементов последовательности от 75-го до 80-го:

```
> Последовательность[75..80] ;
      3077, 9232, 4616, 2308, 1154, 577
```

**Семейство окружностей.** Рассмотрим пример построения графика семейства окружностей. Как образуется это семейство? Берется базовая окружность, делится на равные части, а полученные точки принимаются за центры вычерчиваемых окружностей. Кроме того, все окружности проходят через одну общую точку, взятую на базовой окружности.

Пусть  $R$  — радиус базовой окружности с центром в начале координат. Тогда точки деления могут иметь координаты:

$$\left( R \cos \frac{2\pi}{n} k, R \sin \frac{2\pi}{n} k \right), k = 0, \dots, n-1,$$

где  $n$  — количество точек.

Если в качестве общей точки окружностей взять точку  $(R, 0)$  на оси  $x$ , то радиус  $k$ -й окружности вычисляется по формуле:

$$2R \sin \frac{\pi}{n} k.$$

Для графической иллюстрации примем  $R = 1$  и  $n = 30$ :

```
> restart:
  with(plots): with(plottools):
```

```
> R:= 1: n:= 20:
```

Определим базовую окружность:


```
> gr1:= circle([0,0],R,color=blue,thickness=2,linestyle=3):
```

Сформируем последовательность, содержащую окружности семейства:

```
> f:= k->circle([R*cos(2*Pi*k/n),R*sin(2*Pi*k/n)],
  2*R*sin(Pi*k/n)):
  crcls:= seq(f(k),k=1..n-1):
```

и вычертим все на графике:

```
> display([gr1,crcls],view=[-3.5..2,-3..3]);
```



Кривая, огибающая построенное семейство окружностей, хорошо изучена. Она называется *кардиоидой*.

### 3. ЧИСЛЕННЫЕ МЕТОДЫ РЕШЕНИЯ УРАВНЕНИЙ И ЭКСТРЕМАЛЬНЫХ ЗАДАЧ

#### 3.1. Нахождение минимума функции

**Ситуация 1.** На дачном участке решили построить теплицу площадью  $10 \text{ м}^2$ , затратив при этом как можно меньше материала. Какими должны быть размеры теплицы?

Поразмыслив, приходим к выводу, что расход материала будет зависеть от площади боковых стен, так как площадь крыши теплицы (если она плоская) остается постоянной ( $10 \text{ м}^2$ ) при любых длине и ширине основания.

Пусть длина теплицы равна  $x$  м, ширина —  $y$  м, а высота —  $z$  м. Тогда площадь  $s$  боковых стен (в  $\text{м}^2$ )

$$s = 2xz + 2yz = 2z(x + y).$$

Для решения задачи надо найти наименьшее значение  $s$  при условии, что  $xy = 10$ .

Таким образом, математическая модель построена, осталось ее исследовать.

Сначала отметим, что площадь теплицы не зависит от ее высоты, поэтому  $z$  можно выбрать, исходя из агрономических требований и материальных возможностей.

Следовательно, необходимо найти наименьшее значение  $x + y$  при условии, что  $xy = 10$ . Поскольку  $y = \frac{10}{x}$ , то задача сводится к исследованию на минимум функции одной переменной

$$h(x) = x + \frac{10}{x}.$$

Попробуем найти минимальное значение функции  $h(x)$  приблизительно, построив ее график с помощью Maple.


Прежде всего, определим данную функцию:

```
[> restart:
[> h:= x->x+10/x:
```

Затем построим и вычертим график:

```
[> gr1:= plot(h(x), x=0.5..5, 5..20, color=blue, thickness=2):
```

```
> with(plots) :
  display([gr1], view=[0..5.5, 4..21], axes=boxed,
    tickmarks=[[seq(0..5)], [seq(4..20, 2)]],
    axis=[gridlines=[10, color=green, linestyle=3]]);
```



Из рисунка можно заключить, что  $h(x)$  получает наименьшее значение при  $x \approx 3$ , т. е. приблизительный размер теплицы — 3,0 м x 3,0 м.

Простая проверка полученного результата ( $3 \times 3 = 9 \neq 10$ ) показывает, что определение минимума с помощью графика не является точным.

### Поиск минимума путем просмотра значений

Если большая точность результата не нужна, то минимум функции  $h(x)$  на выбранном отрезке  $[a, b]$  приближенно можно найти, просматривая таблицу ее значений при изменении аргумента  $x$  от  $a$  до  $b$  с заданным шагом  $\varepsilon$ .

Определим из графика границы отрезка:  $a = 2$ ,  $b = 4$ , примем  $\varepsilon = 0,2$ :

```
> restart:
> h:= x->x+10/x:
a:= 2.0 : b:= 4.0 : epsilon:= 0.2:
```

и вычислим таблицу значений  $h(x)$  в виде двух последовательностей:

```
> X:= seq(x,x=a..b,epsilon) : H:= seq(h(x),x=a..b,epsilon) :
  evalf(X,3) ; evalf(H,3) ;

      2.0, 2.2, 2.4, 2.6, 2.8, 3.0, 3.2, 3.4, 3.6, 3.8, 4.0
      7.00, 6.75, 6.57, 6.45, 6.37, 6.33, 6.32, 6.34, 6.38, 6.43, 6.50
```

Функция  $h(x)$  принимает наименьшее значение при  $x = 3,2$ .

### Поиск минимума функции делением пополам

Анализ графика или таблицы значений не очень точный способ нахождения минимума функции. Поэтому рассмотрим численный метод.

Предположим, мы выяснили (с помощью графика или таблицы), что минимум некоторой функции  $f(x)$  находится в пределах отрезка  $[a, b]$ , на котором у нее нет других локальных минимумов. Попытаемся сузить этот отрезок.

Рассмотрим точки  $c$  и  $d$  такие, что  $a < c < d < b$ . Числа  $c$  и  $d$  удобно выбрать вблизи середины отрезка  $[a, b]$  (рис. 3), взяв, например:

$$c = \frac{(a+b)}{2} - \delta, \quad d = \frac{(a+b)}{2} + \delta, \quad \text{где } 0 < \delta < \frac{b-a}{2}.$$




Рис. 3

Заранее нам неизвестно, как расположены точки  $c$  и  $d$  относительно точки минимума функции. Однако, если  $f(c) < f(d)$ , то минимум не может находиться в пределах  $[d, b]$ , и поэтому в качестве правого конца отрезка, содержащего точку минимума, можно взять точку  $d$ . Если же  $f(c) > f(d)$ , то минимум не может находиться в пределах  $[a, c]$ , в этом случае левым концом отрезка, содержащего точку минимума, может быть точка  $c$ . На каждом таком шаге длина выбранного отрезка уменьшается. Когда она становится достаточно малой, можно принять за точку минимума координату середины этого отрезка.

Если параметр  $\delta$  мал, то длина каждого вновь получаемого отрезка примерно вдвое меньше длины предыдущего. Отсюда и название метода.

Составим программу для нахождения наименьшего значения полученной ранее функции  $h(x) = x + \frac{10}{x}$ . В ее основе — цикл с условием. Цикл завершается, когда длина отрезка, содержащего точку минимума, становится меньше заданного значения  $\varepsilon > 2\delta$  (от выбора  $\varepsilon$  зависит точность результата).

Принимаем  $\delta = 0,001$ ,  $\varepsilon = 0,003$ . Определим функцию, введем исходные данные:

```
[> restart:
> h:= x->x+10/x:
a:= 2.0 : b:= 4.0 : delta:= 0.001 : epsilon:= 0.003:
```

Программа для нахождения минимума:

```
[> len:= b-a:
while len>epsilon do
c:= (a+b)/2-delta : d:= (a+b)/2+delta:
if h(c)>=h(d) then a:= c else b:= d end if:
len:= b-a:
end do:
```

Здесь  $len$  — длина отрезка  $[a, b]$ . За точку минимума принимаем середину полученного отрезка:

```
[> minimum:= (a+b)/2 : hmin:= h(minimum) :
evalf(minimum, 4) ;
evalf(hmin, 4) ;

3.162
6.325
```

Итак, функция  $h(x)$  получает наименьшее значение 6,325 при  $x = 3,162$ .

Обратите внимание на предположение о том, что основание теплицы является прямоугольником. Может быть, что при другой форме основания материала на постройку пойдет еще меньше. Однако стоит ли рассматривать такие варианты? Скорее всего нет, поскольку при выборе другого, непрямоугольного основания, возможное сокращение затрат на приобретение материала приведет к возникновению технических сложностей при строительстве.

### Нахождение минимума с помощью встроенной функции

Конечно, в основной библиотеке Maple имеется стандартная функция **minimize**, которая позволяет с большой точностью вычислить наименьшее значение исследуемой функции на указанном отрезке. Поэтому наша задача решается в несколько строк:

```

> restart:
> h:= x->x+10/x:
  hmin:= minimize(h(x), x=2.0..4.0, location):
  evalf(hmin, 5);

```

6.3246, [{x = 3.1623}, 6.3246]}

Вряд ли при строительстве теплицы нужна такая точность.

### 3.2. Движение тела, брошенного под углом к горизонту

Рассмотрим еще одну ситуацию.

**Ситуация 2.** Чтобы наказать непослушную дочь, разгневанный король заточил ее в высокую башню, окруженную глубоким рвом шириной 20 м. В темнице только одно маленькое окошко, находящееся на высоте 8 м от земли. Влюбленный в принцессу юноша хочет послать узнице весть о себе: он думает бросить камень так, чтобы тот попал в окно. Под каким углом к горизонту юноша должен сделать бросок, если начальная скорость, которую он может придать камню, равна 20 м/с?

Будем считать, что размеры камня незначительны по сравнению с размерами башни и рва. Сопротивлением воздуха также пренебрегаем.

Пусть камень, который в начальный момент времени находился в точке  $A$  с координатами  $(x_0, y_0)$ , брошен со скоростью  $v_0$  под углом  $\alpha$  к горизонту (рис. 4).




Рис. 4

Решая задачи о траекториях полета тел, отдельно рассматривают движение тела по горизонтали и по вертикали. Горизонтальная составляющая начальной скорости камня равна  $v_0 \cos \alpha$ , а вертикальная —  $v_0 \sin \alpha$ .

Пусть  $B$  — точка, в которой камень окажется в момент времени  $t$ .

Поскольку мы не учитываем сопротивление воздуха, то движение камня вдоль оси  $x$  является равномерным. Следовательно, горизонтальная составляющая его скорости и координата  $x$  в точке  $B$  определяются соотношениями:

$$v_x = v_0 \cos \alpha, \quad x = x_0 + v_0 t \cos \alpha. \quad (3.1)$$

Из курса физики известно, что движение тела вдоль оси  $y$  — равнопеременное, так как на него действует сила притяжения Земли, направленная вниз. Поэтому вертикальная составляющая скорости и координата  $y$  камня в точке  $B$  определяются соотношениями:

$$v_y = v_0 \sin \alpha - gt, \quad y = y_0 + v_0 t \sin \alpha - \frac{gt^2}{2}, \quad (3.2)$$

где  $g = 9,81 \text{ м/с}^2$  — ускорение свободного падения.

Когда камень достигнет наибольшей высоты подъема, проекция его скорости на ось  $y$  будет равна нулю:  $v_y = 0$ . Отсюда получим время подъема до максимальной высоты

$$t_1 = \frac{v_0 \sin \alpha}{g}. \quad (3.3)$$



Для простоты далее будем считать, что  $x_0 = 0$ ,  $y_0 = 0$ . Подставим выражение (3.3) в уравнение для  $y$  [см. соотношения (3.2)] и получим максимальную высоту подъема камня:

$$h_{\max} = \frac{v_0^2 \sin^2 \alpha}{2g}.$$

Когда камень упадет на землю, координата  $y$  будет равна нулю. Отсюда с учетом выражения (3.3) найдем общее время полета:

$$t_2 = \frac{2v_0 \sin \alpha}{g} = 2t_1. \quad (3.4)$$

Расстояние, которое пролетит камень по горизонтали, т. е. дальность  $L$  полета, определяется подстановкой формулы (3.4) в уравнение для  $x$  [см. соотношения (3.1)]:

$$L = \frac{v_0^2 \sin 2\alpha}{g}. \quad (3.5)$$

Максимальная дальность полета достигается при  $\alpha = 45^\circ$ , поскольку в этом случае  $\sin 2\alpha = 1$ .

Определим теперь траекторию полета камня. Из зависимости для  $x$  [соотношения (3.1)] выразим  $t$  и подставим найденное значение в выражение для  $y$  из соотношений (3.2). В результате получим квадратичную функцию, которая связывает между собой координаты  $x$  и  $y$ :

$$y(x) = x \operatorname{tg} \alpha - \frac{gx^2}{2v_0^2 \cos^2 \alpha}.$$

Графиком этой функции является парабола, направленная ветвями вниз и пересекающая ось  $x$  в точках  $x_1 = 0$  (начало координат) и  $x_2 = L$ .

Вернемся к исходной ситуации. Пусть рост  $y$  юноши выше среднего и координаты начальной точки —  $(0, 2)$ . Конечно, окно в башне имеет какие-то размеры, но для упрощения расчетов предположим, что нужно попасть в его центр, находящийся на высоте 8 м, и, что камень попадет в окно через время  $t$ .

Тогда с учетом того, что  $V_0 = 20$  м/с,  $g = 9,81$  м/с<sup>2</sup>, соотношения (3.1) и (3.2), описывающие движения камня, преобразуются в уравнения:

$$20 = 20t \cos \alpha, \quad 8 = 2 + 20t \sin \alpha - \frac{1}{2} \cdot 9,81t^2.$$

Из первого уравнения найдем  $t = 1/\cos \alpha$ . Подставим  $t$  во второе уравнение и после упрощения получим

$$20 \sin 2\alpha - 12 \cos^2 \alpha - 9,81 = 0 \quad (3.6)$$

Итак, мы должны решить уравнение (3.6) относительно неизвестной величины  $\alpha$  на промежутке изменения ее от 0 до  $\pi/2$ .


### 3.3. Решение нелинейных уравнений

При графической интерпретации корни уравнения вида  $f(x) = 0$  соответствуют точкам пересечения кривой графика функции  $y = f(x)$  с осью  $x$ . Поэтому, если на концах некоторого отрезка  $[a, b]$  непрерывная функция  $f(x)$  принимает значения разных знаков, то внутри этого отрезка уравнение  $f(x) = 0$  имеет хотя бы один корень. Обычно сначала проводится *отделение* корней, т. е. определяются достаточно малые промежутки, содержащие только один корень.

Чтобы выяснить, сколько корней имеет уравнение (3.6) на отрезке  $[0, \pi/2]$ , построим с помощью Maple график функции  $f(\alpha) = 20 \sin 2\alpha - 12 \cos^2 \alpha - 9,81$ :

```
[> restart:
> f:= alpha->20*sin(2*alpha)-12*cos(alpha)^2-9.81;
                                     f:= α→20 sin(2 α) - 12 cos(α)2 - 9.81

[> gr1:= plot(f(alpha), alpha=0..Pi/2, color=blue):
```



Глядя на полученный рисунок, легко локализовать корни: первый находится на отрезке  $[0,4; 0,6]$ , второй — на отрезке  $[1,2; 1,4]$ . Можно отделить корни, построив таблицу значений функции  $f(\alpha)$ , но график — нагляднее.

### Поиск корней уравнения делением пополам

Существуют различные методы нахождения корней.

Пусть корень уравнения  $f(x) = 0$  отделен на некотором отрезке  $[a, b]$ , на концах которого функция  $f(x)$  принимает значения противоположных знаков. Для определенности считаем, что  $f(a) > 0$ . Мы не знаем, какое положение занимает корень относительно точек  $a$  и  $b$ , поэтому делим отрезок  $[a, b]$  пополам точкой  $c = \frac{(a+b)}{2}$ .

Корень находится на той половине отрезка, на концах которой функция  $f(x)$  имеет разные знаки (рис. 5). Новый суженный отрезок снова делим пополам и т. д. На каждом таком шаге длина выбранного отрезка уменьшается вдвое. Когда она окажется достаточно малой, полагаем, что приближенно корень равен координате середины отрезка.




Рис. 5

Составим программу нахождения меньшего из двух корней уравнения (3.6) методом деления пополам. В ее основе — цикл с условием. Цикл завершается, когда длина отрезка, содержащего корень, становится меньше заданного значения  $\varepsilon > 0$  (от выбора  $\varepsilon$  зависит точность результата).

Примем  $\varepsilon = 0,001$ . Так как  $f(0,4) < 0$ ,  $f(0,6) > 0$ , то возьмем  $a = 0,6$  и  $b = 0,4$ :

```
[> a:= 0.6: b:= 0.4: epsilon:= 0.001:
[> len:=abs(b-a):
  while (len > epsilon) do
    c:=(a+b)/2:
    if f(c)>0 then a:=c else b:=c end if:
    len:=abs(b-a):
  end do:
```

Здесь  $len$  — длина отрезка  $[a, b]$ . За приближенное значение корня принимаем координату середины полученного отрезка:

```
[> sol:= (a+b)/2: evalf(sol,3);
0.575
```

или в градусах:

```
[> evalf(sol*180/Pi,3);
33.1
```

Тот же результат получим и с помощью стандартной команды **fsolve**:

```

> fsolve(f(alpha), alpha, alpha=0.4..0.6): evalf(%,3);
0.575

```

Точно также можно вычислить второй корень уравнения. Его приближенное значение равно 1,29 рад или  $73,8^\circ$ .





Рис. 6

Итак, наше исследование показало, что у юноши есть две возможности послать весточку девушке: бросить камень под углом  $33^\circ$  или  $74^\circ$  (рис. 6).

### Задачи для самостоятельного решения

1. Требуется огородить забором прямоугольный участок земли площадью  $295 \text{ м}^2$ , а затем разделить его забором на две равные части. При каких размерах участка (ширине и длине) общая длина забора окажется наименьшей?
2. Какие радиус основания и высоту должен иметь цилиндрический бак, чтобы при объеме  $100 \text{ дм}^3$  на его изготовление пошло наименьшее количество листового металла?
3. Из прямоугольного листа жести размером  $6 \text{ дм} \times 9 \text{ дм}$  нужно изготовить коробку. В четырех углах листа вырезают одинаковые квадраты и загибают края под прямым углом (см. рисунок 7). Какова наибольшая вместимость такой коробки?

У к а з а н и е. Решение задачи сводится к исследованию функции, выражающей зависимость объема от параметра  $x$ .



4. Пушка выстреливает ядро со скоростью  $v_0 = 100$  м/с. Ядро падает на расстоянии  $L = 1000$  м от пушки. Под каким углом  $\alpha$  к горизонту сделан выстрел? Сопротивлением воздуха пренебречь.

У к а з а н и е. Записать формулы (3.5) в виде:  $v_0^2 \sin 2\alpha - gL = 0$ . Затем решить полученное нелинейное уравнение относительно  $\alpha$ .

5. Тело брошено вдоль склона вниз под углом  $\alpha$  к поверхности горы. Определить дальность полета, если начальная скорость равна  $v_0$ , угол наклона горы  $\beta$ . Сопротивлением воздуха пренебречь.

6. Пушка выстреливает ядро под углом  $\alpha = 60^\circ$  к горизонту со скоростью  $v = 100$  м/с. Когда ядро достигает наивысшей точки траектории, пушка стреляет еще раз. Через какое время после первого выстрела ядра окажутся на минимальном расстоянии друг от друга (пока оба они в полете)? Чему равно это расстояние?

## 4. МОДЕЛИРОВАНИЕ В ЗАДАЧАХ ПРОЦЕНТНОГО РОСТА

### 4.1. Задачи на проценты

Понятие «процент» широко используется в экономике, торговле, статистике, социологии, в различных отраслях науки и техники. Задачи, связанные с процентами, особенно часто встречаются в банковском деле.

**Процентом** некоторой величины называется сотая часть этой величины. Процент обозначается знаком %.

При решении задач на проценты встречаются следующие три операции.

**Нахождение процента от величины.** Задана величина  $A$ . Величина  $X$  составляет  $k$  % от  $A$ . Найти  $X$ .

Разделив  $A$  на 100, найдем 1 % от  $A$ . Умножив результат на  $k$ , получим  $k$  % от величины  $A$ . Таким образом,

$$X = \frac{A \cdot k}{100}. \quad (4.1)$$

**Нахождение процентного значения одной величины по отношению к другой.** Заданы  $X$  и  $A$ . Найти число  $k$  процентов, которые  $X$  составляет от  $A$ .

Из формулы (4.1) определим

$$k = \frac{X \cdot 100 \%}{A}. \quad (4.2)$$

Например, если  $X = 0,5A$ , то  $X$  составляет 50 % от  $A$ , если  $X = 1,5A$ , то  $X$  составляет 150 % от  $A$ .

**Нахождение величины по известному проценту ее части.** Известна величина  $X$ , которая составляет  $k$  % от  $A$ . Найти  $A$ .

Разделив  $X$  на  $k$ , получим 1 % от  $A$ . Умножив результат на 100, найдем  $A$ :

$$A = \frac{X \cdot 100}{k}. \quad (4.3)$$

Пусть, например,  $X = 15$ , что составляет 20 % от  $A$ . Тогда согласно формуле (4.3)  $A = 75$ .

Заметим, что операции (4.1) и (4.3) являются обратными по отношению друг к другу.

Рассмотрим несколько простых примеров.

**Пример 1.** После снижения цен товар, стоивший 150 000 руб., стал стоить 120 000

руб. На сколько процентов были снижены цены?

Так как начальная цена  $X$  принимается за 100 %, то новая цена  $A$  составляет  $\frac{120000 \cdot 100}{150000} = 80\%$  от старой. Следовательно, цена была снижена на 20 %.

**Пример 2.** Жирность молока 2,5 %. Сколько жира в 1 л молока? (Плотность молока принять равной 1 г/см<sup>3</sup>.)

Количество жира в 1 л, т. е. 1000 г, молока рассчитаем по формуле (4.1):

$$\frac{1000 \cdot 2,5}{100} = 25 \text{ (г)}.$$

**Пример 3.** Известно, что в руде 40 % примесей. Сколько необходимо взять руды, чтобы получить 300 кг чистого металла?

В руде 60 % чистого металла, масса которого после удаления примесей 300 кг. Поэтому согласно формуле (4.3) масса руды:

$$\frac{300 \cdot 100}{60} = 500 \text{ (кг)}.$$

Рассмотрим более сложные примеры.

**Пример 4.** Цену на товар снизили сначала на 20 %, затем на 10 %, а потом на 5 %. Как изменилась цена товара по сравнению с начальной?

Для удобства вычислений будем считать, что начальная цена товара  $100A$  руб. Тогда после первого снижения цена товара составила 80 % от  $100A$ , т. е.  $80A$  руб., после второго — 90 % от  $80A$  руб., т. е. согласно формуле (4.1)  $\frac{80A \cdot 90}{100} = 72A$  руб. После третьего

снижения цен стоимость товара стала равной 95 % от  $72A$  руб., т. е.  $\frac{72A \cdot 95}{100} = 68,4A$  руб.

Таким образом, после трех уценок первоначальная цена товара снизилась на 31,6A руб., что составляет 31,6 % от  $100A$  руб.

**Пример 5.** Цену на товар сначала снизили на 10 %, а затем повысили на 10 %. Как изменилась цена товара по сравнению с первоначальной?

Пусть начальная цена товара  $100A$  руб. После уценки новая цена товара составляла 90 % от начальной, т. е.  $90A$  руб., после повышения — 110 % от  $90A$  руб., т. е.



$$\frac{90A \cdot 110}{100} = 99A \text{ руб.}$$

Следовательно, цена товара по сравнению с первоначальной понизилась на 1 %.

**Пример 6.** В свежих яблоках 90 % воды, в сухих — 10 %. Сколько сухих яблок получается из 200 кг свежих? На сколько процентов изменяется масса яблок при сушке?

В свежих яблоках 10 % сухого вещества, т. е.  $\frac{200 \cdot 10}{100} = 20$  кг. Эти 20 кг составляют

80 % от массы свежих яблок. Воспользовавшись формулой (4.1) рассчитаем массу сухих

яблок  $\frac{20 \cdot 100}{80} = 25$  кг. Это согласно формуле (4.2) составляет  $\frac{25 \cdot 100}{200} = 12,5$  % от мас-

сы свежих яблок.

При моделировании с использованием процентов не имеет значения, что исследуется: химический состав веществ, цены на товары или количество объектов с определенными свойствами, учитываются только *соотношения* между различными величинами.

*На всех этапах исследования ситуации за 100 % принимают значения величин, которыми оперируют на этих этапах.*

#### 4.2. Простые и сложные проценты

Вы открыли счет в банке и внесли некоторую сумму, называемую *основной суммой* или *капиталом*. Если вклад вносится на определенное время — *период*, то по истечении этого времени доход от капитала, называемый *процентными деньгами* или *процентами*, присоединяется к основной сумме. Если вклад вносится на неопределенное время, то доход от капитала начисляется все время, но способы начисления зависят от договоренности с банком.

Отношение процента к основной сумме называется *нормой процента*. Эта величина чаще всего выражается в процентах. Например, вклад 1 000 000 руб. при норме процента 20 % через год составит 1 200 000 руб. (1 000 000 руб. вложенных + 200 000 руб. процентов).

По сходным правилам могут расти не только банковские вклады, но и другие величины, например, численность народонаселения. Поэтому рассмотрим ситуацию с финансами более подробно.

Пусть  $P_0$  — основная сумма банковского вклада,  $r$  — норма процента за один период

(например, месяц, квартал или год). Процент, который начисляется только на основную сумму и выплачивается в конце периода, называется *простым процентом*. В этом случае вклад всегда увеличивается на сумму  $\frac{P_0 r}{100}$ , т. е. в конце первого периода он будет

$$P_1 = P_0 + \frac{P_0 r}{100} = P_0 \left( 1 + \frac{r}{100} \right),$$

спустя два периода

$$P_2 = P_1 + \frac{P_0 r}{100} = P_0 \left( 1 + \frac{r}{100} \right) + \frac{P_0 r}{100} = P_0 \left( 1 + \frac{2r}{100} \right) \text{ и т. д.}$$

Через  $n$  периодов времени вклад возрастет до величины

$$P_n = P_0 \left( 1 + \frac{nr}{100} \right). \quad (4.4)$$

Отсюда получаем следующие соотношения:

$$P_0 = \frac{P_n}{\left( 1 + \frac{nr}{100} \right)}, \quad n = \frac{100 (P_n - P_0)}{r P_0}, \quad r = \frac{100 (P_n - P_0)}{n P_0}. \quad (4.5)$$

По формулам (4.5), зная три заданные величины, находим значение неизвестной четвертой.

**Пример 7.** Какую сумму денег надо положить в банк, чтобы через пять лет получить 3 000 000 руб., если простой процент, выплачиваемый банком, равен 10 %?

Начальный вклад вычислим по первой из формул (4.5):

```
[ > P0:=3*10^6/(1+5*10/100): evalf(P0,2);
                               2.0 10^6
```

Следовательно, в банк нужно положить 2 000 000 руб.

**Пример 8.** На счет в банк, который выплачивает простой процент — 15 %, положили 1 000 000 руб. Через какое время на счете будет 2 000 000 руб.?

Время вычислим по второй из формул (4.5):

```
[ > n:=100*(2*10^6-10^6)/(15*10^6): evalf(n,2);
                               6.7
```

То есть время получения ожидаемой прибыли приблизительно 7 лет.

На практике чаще всего дается годовая норма  $r$  процента. Если по соглашению с банком можно снять вклад (не теряя прибыли) менее чем через год, то процент, выплачиваемый за месяц, как правило, определяется из расчета  $r/12$ , а выплачиваемый за день —  $r/365$ .

**Пример 9.** Вы взяли займы на год 1 000 000 руб. при годовом простом проценте 15 %, а через три месяца решили рассчитаться с кредитором. Какую сумму Вы должны будете вернуть?

Простой процент, выплачиваемый за месяц, равен  $15/12 = 5/4$ . Возвращаемую сумму вычислим по формуле (4.4):

$$\left[ \begin{array}{l} > 1000000 + 1000000 * 3 * (5/4) / 100; \\ \\ \end{array} \right. \quad 1037500$$

Следовательно, кредитору Вы должны вернуть 1 037 500 руб.

Приведенные примеры показывают, что вклады под простые проценты могут быть выгодны для вкладчиков лишь в том случае, когда вклад вносится на небольшое время и под большой процент. Гораздо более привлекательны вклады под **сложные проценты**, поскольку сумма, с которой начисляются проценты, в этом случае постоянно растет.

Сущность расчетов по сложным процентам заключается в том, что проценты, начисленные в текущем периоде, в следующем периоде присоединятся к основной сумме, в результате проценты будут начислены и на основную сумму, и на деньги, добавленные за предыдущий период.

Предположим в банк положили  $P_0$  руб., и  $r$  — норма сложного процента за один период (месяц, квартал, год). Тогда вклад в конце первого периода будет

$$P_1 = P_0 + \frac{P_0 r}{100} = P_0 \left( 1 + \frac{r}{100} \right),$$

спустя два периода

$$P_2 = P_1 + \frac{P_1 r}{100} = P_1 \left( 1 + \frac{r}{100} \right) = P_0 \left( 1 + \frac{r}{100} \right) \left( 1 + \frac{r}{100} \right) = P_0 \left( 1 + \frac{r}{100} \right)^2 \text{ и т. д.}$$

Через  $n$  периодов времени вклад возрастет до величины

$$P_n = P_0 \left(1 + \frac{r}{100}\right)^n. \quad (4.6)$$

Полученное равенство называется *формулой сложного процента*.

Заметим, что для  $n = 1$  формулы (4.4) и (4.6) эквивалентны.

Из формулы (4.6) получим следующие соотношения:

$$P_0 = \frac{P}{\left(1 + \frac{r}{100}\right)^n}, \quad n = \frac{\ln P - \ln P_0}{\ln \left(1 + \frac{r}{100}\right)}, \quad r = 100 \left[ \left(\frac{P}{P_0}\right)^{\frac{1}{n}} - 1 \right]. \quad (4.7).$$

По формулам (4.7), зная три заданные величины, находим значение неизвестной четвертой.

**Пример 10.** У Вас есть выбор: положить 1 000 000 руб. на вклад с простым процентом (при норме 12 %) или сложным (при норме 10 %). Какой из вкладов выгоднее?

Рост вклада в течение, например, шести ближайших периодов вычислим по формулам (4.4) и (4.6):

```

> restart :
> СложныйПроцент := (P0, r, n) -> P0 * (1+r/100) ^ n;
ПростойПроцент := (P0, r, n) -> P0 * (1+(r/100) * n) ;

СложныйПроцент := (P0, r, n) -> P0 * (1 + (r/100) ^ n)
ПростойПроцент := (P0, r, n) -> P0 * (1 + (r/100) * n)

> seq(СложныйПроцент(1, 10, i), i=1..6) :
evalf(%, 8) ;

1.1000000, 1.2100000, 1.3310000, 1.4641000, 1.6105100, 1.7715610


> seq(ПростойПроцент(1, 12, i), i=1..6) :
evalf(%, 8) ;

1.1200000, 1.2400000, 1.3600000, 1.4800000, 1.6000000, 1.7200000

```

Видим, что первые четыре периода большую прибыль приносит вклад с простым процентом, а начиная с пятого — со сложным.

Для наглядности построим с помощью Maple графики функций, которые определяют простой (12 %) и сложный (10 %) проценты:



Анализ графиков позволяет сделать тот же вывод.

**Пример 11.** Сумма в размере 5 000 долларов взята в кредит на 5 лет по ставке сложного процента, равной 10 % годовых. Определить проценты и сумму, подлежащую возврату.

Сумму определим по формуле (4.6):

```

> s:=5000*(1+10/100)^5: evalf(s,6);
      8052.55

```

Таким образом, через 5 лет необходимо вернуть 8 052 доллара, из этой суммы 5 000 долларов — это долг, а 3 052 доллара — проценты.

**Пример 12.** Остров Манхэттен, на котором расположена центральная часть Нью-Йорка, был выменян у индейцев на товары ценой 24 доллара. Стоимость земли этого острова через 350 лет оценивалась примерно в 40 миллиардов долларов. Под какой сложный процент 350 лет назад следовало бы поместить в банк 24 доллара, чтобы сейчас выкупить землю острова?

Найдем ответ, решив уравнение (4.6) относительно  $r$ :

```

> fsolve(24*(1+r/100)^350=40*10^9,r);
-206.2546975, 6.254697515

```

Отбросив отрицательное значение, заключаем, что норма процента примерно 6,25 % годовых.

Сложный процент может начисляться чаще, чем раз в год, например, раз в полгода, квартал или месяц. При годовой норме  $r$  процент, выплачиваемый за месяц, как правило, считается равным  $r/12$ , выплачиваемый за квартал —  $r/4$ , за день —  $r/365$ .

**Пример 13.** В банк принят вклад 50 000 руб. сроком на 90 дней по ставке 10,5 % годовых с начислением процентов каждые 30 дней. Определить сумму вклада через три месяца.

Согласно условию имеем  $90/30 = 3$  периода, длительность периода дана в днях, поэтому для определения процента, выплачиваемого каждые 30 дней, годовая норма процента делится на 365 и умножается на 30. Сумму вклада через три периода рассчитаем по формуле (4.6):

```

> s:=5*10^4*(1+(10.5/365*30)/100)^3: evalf(s,7);
51305.72

```

**Пример 14.** Норма сложного процента равна 4, 12, 16 и 20 % годовых. Как будет расти вклад, если его начальная сумма равна 1000 долларов?

Определим функцию сложных процентов:

```

> P:=(P0,r,n)->P0*(1+r/100)^n;
P:=(P0,r,n)->P0*(1+1/100*r)^n

```


Ответ представим в виде графиков зависимости  $P$  от  $n$ :

```

> P0:= 1000: with(plots):
gr1:= plot(P(P0,4,n), n=0..10, color=blue):
gr2:= plot(P(P0,12,n), n=0..10, color=blue):
gr3:= plot(P(P0,16,n), n=0..10, color=blue):
gr4:= plot(P(P0,20,n), n=0..10, color=blue):
gr5:= textplot([8.5,5500,"20%"], font=[COURIER,12]):
gr6:= textplot([8.5,4000,"16%"], font=[COURIER,12]):
gr7:= textplot([8.5,3000,"12%"], font=[COURIER,12]):
gr8:= textplot([8.5,1600,"4%"], font=[COURIER,12]):

```

```
> display([seq(gr||i,i=1..8)],view=[0..10.1,10^3..6.5*10^3],
axes=boxed, axesfont=[COURIER,10], thickness=2,
axis=[gridlines=[10,color=green]]);
```



### Непрерывное начисление процентов

На практике иногда проценты начисляются *непрерывно*, за сколь угодно малый промежуток времени. В этом случае через  $n$  периодов

$$P_n = P_0 e^{nr/100}, \quad (4.8)$$

где  $e \approx 2,71828$  — основание натуральных логарифмов. Эту формулу приводим без вывода.

Непрерывное начисление процентов используется при анализе сложных финансовых задач, например, обоснование и выбор инвестиционных решений. Оценивая работу финансового учреждения, куда платежи в течение одного периода поступают многократно, целесообразно предполагать, что наращенная сумма непрерывно меняется во времени и применять непрерывное начисление процентов.

**Пример 15.** Кредит в размере 10 000 долларов получен сроком на 2 года под 8 % годовых. Определить сумму, подлежащую возврату, если проценты будут начисляться: а) один раз в год; б) один раз в квартал; в) ежедневно; г) непрерывно.

а) Согласно условию имеем два годовых периода, используем формулу (4.6) и считаем искомую сумму:

```
[ > s:=10^4*(1+0.08)^2: evalf(s,7);
                               11664.00
```

Получили, что следует вернуть 11 664 доллара.

б) В году четыре квартала, поэтому годовая норма процента делится на 4, а всего имеем  $2 \cdot 4 = 8$  периодов, длительность каждого — один квартал. Рассчитаем искомую сумму:

```
[ > s:=10^4*(1+0.08/4)^(2*4): evalf(s,7);
                               11716.59
```

в) При ежедневном начислении процентов имеем  $2 \cdot 365$  периодов, а годовая норма процента делится на 365. Тогда получим:

```
[ > s:=10^4*(1+0.08/365)^(2*365): evalf(s,7);
                               11734.90
```

г) Проценты начисляются непрерывно, поэтому для расчета используем формулу (4.8):

```
[ > s:=10^4*exp(0.08*2): evalf(s,7);
                               11735.11
```

Следует отметить, что это наиболее невыгодные условия кредитования.

### Задачи для самостоятельного решения

1. Рыночная цена на картофель повысилась на 20 %, а через некоторое время понизилась на 20 %. Когда картофель стоил дороже: до повышения или после снижения цены?

2. Цену на товар увеличили сначала на 10, а затем на 15 %. Подсчитайте общий процент повышения цены. Был бы таким же общий процент, если бы цену сначала увеличили на 15, а потом на 10 %? А если бы первый раз цену повысили на 20 %, а затем на 5 %? А если бы цену сразу увеличили на 25 %?

3. Цена входного билета на стадион была 10 000 руб. После ее снижения число зрителей возросло на 25 %, а выручка увеличилась на 12,5 %. Сколько стал стоить билет после снижения цены?



4. Товар стоил 3000 руб. Его 2 раза уценили, оба раза снизив цену на одно и то же число процентов, в результате он стал стоить 2430 руб. На сколько (в процентах) уценили товар в первый и во второй раз?

5. Имеются два сплава меди и цинка с содержанием меди 20 и 40 % соответственно. В каком отношении нужно взять эти сплавы, чтобы переплавив взятые куски вместе, получить сплав, содержащий 25 % меди?

6. Аквариум частично заполнен водой. За месяц 40 % воды испарилось, при этом объем воздуха в аквариуме увеличился на 60 %. Какую часть аквариума (в процентах) занимала вода в конце месяца?

7. Содержание воды в свежих грибах составляет 90 %, а в сухих — 10 %. Сколько сухих грибов получится из 30 кг свежих?

8. В городе проживает 1 000 000 человек, каждый год число жителей увеличивается на 3 %. Каким оно окажется через 5 лет?

9. Через 6 лет для покупки дачи Вам потребуется 20 000 000 руб. Какую сумму необходимо положить в банк сейчас с нормой сложных процентов — 12 % в год, чтобы через 6 лет можно было купить эту дачу?

10. Себестоимость изделия понизилась за первое полугодие на 10 %, а за второе — на 20 %. Определить его первоначальную себестоимость, если новая себестоимость 72 000 руб.

11. Банк выплачивает 12 % годовых, причем начисляются сложные проценты. Через сколько лет внесенная сумма удвоится?

12. Что выгоднее для вкладчика: забрать 200 000 руб., которые есть у него на счете, или получить 350 000 руб. через 3 года, если процентная ставка сложных процентов равна 17 %.

## 5. БИОЛОГИЧЕСКИЕ И ЭКОЛОГИЧЕСКИЕ МОДЕЛИ

### 5.1. Модель Мальтуса

Рассмотрим некоторый биологический вид, у которого нет естественных врагов и достаточно корма. Обозначим численность особей вида  $x$ . Будем считать, что ее увеличение за определенный заданный период времени пропорционально числу особей в начале периода, т. е. равно  $ax$ , где  $a > 0$  — коэффициент прироста. Обозначим через  $x_n$  число особей в популяции<sup>1</sup> спустя  $n$  периодов времени. Тогда

$$x_n = x_{n-1} + ax_{n-1} = (1 + a)x_{n-1}, n = 1, 2, \dots \quad (5.1)$$

Если внимательно проанализировать ситуацию, то станет очевидно, что  $x_n$  можно определить по формуле, аналогичной формуле (4.6) сложных процентов:

$$x_n = (1 + a)^n x_0,$$

где  $x_0$  — численность популяции в начальный момент времени.

**Пример 1.** Пусть на изолированной территории обитает 100 особей некоторого вида. Каждый год их количество увеличивается на 5 %. Определить численность вида через три года.

Согласно условию  $a = 0,05$ . Воспользуемся формулой (5.1) и найдем (каждый раз берем округленное целое значение) численность вида через три года:

```
> restart:
> x0:= 100: a:= 0.05:
  x1:=round((1+a)*x0): x2:=round((1+a)*x1): x3:=round((1+a)*x2):
  x0,x1,x2,x3;

                                100, 105, 110, 116
```

Через три года на территории будут обитать 116 особей.

Это — простейшая модель развития популяции. Ее легко реализовать в виде программы, которая для заданных значений  $x_0$  и  $a$  создает таблицу значений  $x$ . Для хранения таблицы используется массив.

Вначале зададим начальные значения параметров и размер таблицы:

---

<sup>1</sup> Популяция — совокупность особей одного вида, длительно занимающая определенное пространство и воспроизводящая себя в течение большого числа поколений.

```
[> restart:
> a:= 0.05: x0:= 100: n:= 15: X:= array(0..n):
```

С помощью цикла вычислим нужное количество значений:

```
[> X[0]:= x0:
  for i to n do
    X[i]:= round((1+a)*X[i-1])
  end do:
```

Численность популяции в конце каждого из периодов:

```
[> Таблица:= seq(X[i],i=1..n):
  seq(Таблица[i],i=1..n);

      105, 110, 116, 122, 128, 134, 141, 148, 155, 163, 171, 180, 189, 198, 208
```

Определим наименьшее и наибольшее значения:


```
[> min(Таблица),max(Таблица);

      105, 208
```

Затем построим график:

```
[> with(plots):
  gr1:= listplot([Таблица],color=blue,thickness=2):
  gr2:= listplot([Таблица],color=red,
  style=point,symbol=circle,symbolsize=15):
```

```
> display([gr1, gr2], view=[0..n+1, 100..210],
          title="Динамика популяции", axes=boxed,
          axesfont=[COURIER, 12], titlefont=[COURIER, 13],
          axis=[gridlines=[15, color=green, linestyle=3]]);
```



Понятно, что рассматриваемая модель является весьма упрощенной. В действительности рост численности популяции ограничен различными факторами: наличием естественных врагов, имеющейся кормовой базой, эпидемиями и т. п. Однако в некоторых случаях, например, при исследовании размножения бактерий в лабораторных условиях, эта модель оказывается вполне приемлемой.

Впервые рассматриваемая модель была предложена английским экономистом Т. Р. Мальтусом (1766—1834) в конце XVIII века. На ее основе он сформулировал положение о том, что население Земли будет расти в геометрической прогрессии, несмотря на «естественное» регулирование в периоды голода, эпидемий и войн, а средства существования — в арифметической. То есть ученый предсказывал нехватку средств существования, в частности, продовольствия. Мальтус не мог предвидеть появления новых технологий, увеличивающих производство продуктов питания, поэтому его мрачные предсказания пока не оправдываются.

## 5.2. Загрязнение реки

**Пример 2.** В результате аварийного сброса промышленных стоков концентрация  $c_0$  вредных веществ в реке сильно увеличилась. Через сутки после аварии уровень загрязнения снизился и стал равным  $c_1 = kc_0$ , где  $k$  — коэффициент естественного самоочищения,  $0 < k < 1$ . Предположим, что дальнейшее самоочищение реки будет проходить так же, как и в первые сутки. Тогда через двое суток  $c_2 = kc_1 = k^2c_0$ . Каким будет уровень загрязнения реки вредными веществами (см. таблицу) через сутки, двое и в последующие дни, до тех пор, пока их концентрация не станет меньше предельно допустимой  $d$ ?

Провести исследование для веществ, указанных в следующей таблице.

Таблица

**Вредные вещества, попавшие в реку**

Вещество	$c_0$	$d$	$k$
Свинец	2,5	0,03	0,83
Мышьяк	1,5	0,05	0,95
Фтор	0,2	0,05	0,98

*Примечание.* Коэффициент  $k$  определяется на основании изучения последствий аналогичных аварий, имевших место в прошлом, и зависит от скорости течения реки, состояния ее берегов, наличия водорослей и других факторов.

Ограничимся реализацией модели для свинца. Определим исходные данные:

```
[> restart:
> НачальнаяКонцентрация:= 2.5: ПредельнаяКонцентрация:= 0.03:
  КоэффициентУменьшения:= 0.83:
```

Построим таблицу, которая отражает уровень загрязнения реки по суткам. Таблица формируется в виде списка:

```
[> УровеньЗагрязнения:= НачальнаяКонцентрация:
  Свинец:= НачальнаяКонцентрация:
  Сутки:= 1:

> while Свинец > ПредельнаяКонцентрация do
  Сутки:= Сутки+1:
  Свинец:= КоэффициентУменьшения*Свинец:
  УровеньЗагрязнения:= УровеньЗагрязнения,Свинец:
end do:
```


```
[ > Сутки;
                                     25
```

Таким образом, загрязнение реки свинцом снизится до допустимых пределов через 25 суток. Ход изменения концентрации свинца в течение этого времени:

```
[ > Таблица := [УровеньЗагрязнения] :
[ > evalf(Таблица [1..12], 2) ; evalf(Таблица [13..25], 2) ;
                                     [2.5, 2.1, 1.7, 1.4, 1.2, 0.98, 0.82, 0.68, 0.56, 0.47, 0.39, 0.32]
                                     [0.27, 0.22, 0.18, 0.15, 0.13, 0.11, 0.087, 0.073, 0.060, 0.050, 0.041, 0.034, 0.029]
```

Для большей наглядности полученных результатов покажем их на графике изменения концентрации свинца от времени:

```
[ > with(plots) : with(plottools) :
  gr1 := listplot(Таблица, color=blue, thickness=2) :
[ > display([gr1], view=[0..Сутки+1, -0.2..2.6],
  xtickmarks=[seq(0..Сутки, 2)], ytickmarks=[seq(0..3, 0.5)],
  axis=[gridlines=[10, color=green, linestyle=3]]) ;
```



Экологические модели позволяют прогнозировать изменения состояния различных экосистем в течение некоторого времени и в случае необходимости принимать меры по улучшению природопользования и охраны живых организмов.

### 5.3. Модели ограниченного роста

Уточним модель Мальтуса, сделав ее более реалистичной. Попробуем учесть некоторые факторы, препятствующие росту численности популяции.

Если животные обитают на ограниченной территории, то с увеличением их числа неизбежной становится конкуренция за жизненное пространство и корм. Встречи особей друг с другом происходят чаще, что приводит к распространению болезней и росту смертности, т. е. сокращению (убыванию) численности популяции. Понятно, что убыль, зависящая от частоты встреч, пропорциональна их количеству, т. е. равна  $bx^2$ , где  $b > 0$  — коэффициент убывания. Поэтому будем считать, что

$$x_n = x_{n-1} + ax_{n-1} - bx_{n-1}^2 = (1 + a - bx_{n-1})x_{n-1}, \quad n = 1, 2, \dots \quad (5.2)$$

Эта уточненная модель называется *моделью ограниченного роста*.

Таблица изменения численности популяции получается точно так же, как и для исходной модели Мальтуса (см. 5.1). Проведем вычисления и построим графики для различных значений параметра  $a$  и  $b = 0,0001$ .

Для  $a = 0,05$  численность популяции сначала быстро растет, а по достижении численности порядка 500 особей практически не изменяется (рис. 7):




Рис. 7

Для  $a = 0,01$  прирост и убыль животных уравниваются, и численность популяции стабилизируется (рис. 8, а), для  $a = 0,009$  она довольно быстро уменьшается (рис. 8, б).




Рис. 8

Немного изменим рассматриваемую ситуацию.

**Пример 3.** Для производства вакцины на заводе выращивают одну из культур бактерий. Известно, что если масса бактерий (в граммах) —  $x$ , то через день она возрастает на величину  $(a - bx)x$ , где  $a$  и  $b$  — коэффициенты, зависящие от вида бактерий.

Ежедневно для нужд производства отбирают  $m$  г бактерий, т. е. масса оставшихся бактерий определяется следующей формулой:

$$x_n = (1 + a - bx_{n-1})x_{n-1} - m, \quad n = 1, 2, \dots \quad (5.3)$$

Эта модель называется *моделью ограниченного роста с отбором*. Приемы ее реализации в Maple аналогичны уже рассмотренным.

#### 5.4. Модель «жертвы и хищники»

Моделируя развитие популяции, мы не учитывали того, что у животных могут быть враги.

Пусть на некоторой территории обитают два вида: *жертвы*, питающиеся подножным кормом, и *хищники*, охотящиеся на жертв. В качестве пары жертва — хищник могут выступать волки и овцы, щуки и караси, лисы и зайцы.

Обозначим численность жертв и хищников соответственно через  $x$  и  $y$ . Жертвы при отсутствии хищников размножаются с коэффициентом прироста  $a$ , а хищники без добычи вымирают с коэффициентом убывания  $b$ .

Полагаем, что численность жертв из-за встреч с хищниками уменьшается пропорци-



онально количеству тех и других с коэффициентом  $r > 0$ , а численность хищников соответственно увеличивается с коэффициентом  $s > 0$ . Тогда

$$x_n = x_{n-1} + ax_{n-1} - rx_{n-1}y_{n-1} = (1 + a - ry_{n-1})x_{n-1}, \quad (5.4)$$

$$y_n = y_{n-1} - by_{n-1} + sx_{n-1}y_{n-1} = (1 - b + sx_{n-1})y_{n-1}.$$

Впервые модель «хищники и жертвы» рассмотрел итальянский математик В. Вольтерра (1860—1940) в 1931 году.

Обозначим через  $x_0$  и  $y_0$  количество животных каждого вида в начальный момент времени. Построим для заданных значений параметров таблицы значений  $x$  и  $y$  и графики.

Определим начальные данные:


```
[> restart :
> a := 0.035 : r := 0.0005 : b := 0.015 : s := 0.0002 :
  x0 := 100 : y0 := 40 :
```

Программа вычисления таблиц значений  $x$  и  $y$ , рассчитанных по формулам (5.4), для  $n = 800$ :

```
[> n := 800 : X := array(0..n) : Y := array(0..n) :
  X[0] := x0 : Y[0] := y0 :
  for i from 1 to n do
    X[i] := (1 + a - r·Y[i-1])·X[i-1];
    Y[i] := (1 - b + s·X[i-1])·Y[i-1];
  end do :
> Жертвы := seq(X[i], i = 1..n) : Хищники := seq(Y[i], i = 1..n) :
```

По табличным данным построим графики изменения численности жертв и хищников:

```
[> with(plots) :
  gr1 := listplot([Жертвы], color = blue, linestyle = 3) :
  gr2 := listplot([Хищники], color = red) :
  display([gr1, gr2], view = [0..n, 0..180], axes = boxed,
    axesfont = [COURIER, 10], thickness = 2,
    axis = [gridlines = [15, color = green]]);
```



На полученном рисунке хорошо видна периодическая зависимость числа хищников (сплошная линия) и жертв (пунктирная линия) от времени.

### 5.5. Модель Лесли

Теперь рассмотрим математическую модель происходящих со временем изменений численности некоторого биологического вида, предложенную в 1948 году американским биологом П. Лесли.

Распределим особей по возрастным группам: в первую группу попадают все особи в возрасте до года, во вторую — все не попавшие в первую группу, возраст которых меньше двух лет, и т. д. Предполагается, что всего будет  $n$  таких групп.

Пусть в  $i$ -ю возрастную группу входит  $p_i$  особей. Тогда общее число особей в популяции выражается суммой

$$p_1 + p_2 + \dots + p_n.$$

У каждой возрастной группы свой коэффициент рождаемости —  $b_i$ . Это означает, что годовой прирост  $i$ -й группы с численностью  $p_i$  будет равен  $b_i p_i$ . Таким образом, годовой прирост всей популяции равен сумме

$$b_1 p_1 + b_2 p_2 + \dots + b_n p_n.$$

Все эти особи в следующем году попадут в первую возрастную группу.

Конечно, в каждой группе возможна убыль, вызванная болезнями, несчастными случаями, нападением хищников и т. п. Поэтому наряду с коэффициентом рождаемости для каждой  $i$ -й группы вводится и коэффициент выживания —  $s_i$  ( $0 \leq s_i \leq 1$ ). Это означает, что из  $p_i$  особей  $i$ -й группы через год остается в живых  $s_i p_i$ . Считается, что все особи по-

следней группы в течение года погибают, т. е.  $s_n = 0$ .

Таким образом, через год получим следующее распределение популяции по годовым возрастным группам:

$$b_1 p_1 + b_2 p_2 + \dots + b_n p_n, s_1 p_1, \dots, s_{n-1} p_{n-1}.$$

Коэффициенты выживания, так же как и коэффициенты рождаемости, выводятся на основе многолетних наблюдений за популяциями данного вида. Отмечено, например, что для ящериц характерна слабая зависимость  $s_i$  от  $i$ , т. е. выполняются приближенные равенства  $s_1 = s_2 = \dots = s_n$ ; убывание  $s_i$  с возрастанием  $i$  характерно для слонов и китов; максимум  $s_i$  для срединных значений  $i$  отмечается у пингвинов; крайне малое значение  $s_i$  для начальных значений  $i$  наблюдается у некоторых видов рыб.

Пусть дано  $n$  и известны  $p_1, p_2, \dots, p_n$  — первоначально зарегистрированное распределение особей по возрастным группам;  $b_1, b_2, \dots, b_n$  — коэффициенты рождаемости;  $s_1, s_2, \dots, s_n$  — коэффициенты выживания. Как будет изменяться численность  $t_1, t_2, \dots, t_m$  популяции в течение  $m$  лет?

Пусть исходные параметры заданы следующими списками:

```
> n := 10 : m := 10 :
  P := [12, 14, 15, 20, 24, 19, 15, 10, 6, 3] :
  B := [0, 0, 0.5, 1.2, 1.0, 0.5, 0.2, 0.1, 0, 0] :
  S := [0.8, 0.85, 0.9, 0.95, 0.99, 0.99, 0.75, 0.5, 0.1, 0] :
```

Примем, что количество особей, вычисляемое ежегодно, будет сохраняться в массиве

$T$ . Начальная численность популяции:

```
> T := array(0..m) :
  T[0] := sum(P[i], i=1..n) : T[0] ;
```

138

Тогда программа вычисления значений  $t_1, t_2, \dots, t_m$ :

```

> for i to m do
  u:= 0;
  for j from n by -1 to 2 do
    u:= u+P [j] *B [j]:
    P [j] := P [j-1] *S [j-1]:
  end do;
  P [1] := u+P [1] *B [1];
  T [i] := sum(P [k] ,k=1..n)
end do:

```

Учитываем, что значения  $p_j$  нужно заменять на  $p_{j-1}s_{j-1}$  в порядке убывания индекса: сначала заменяем  $p_n$ , затем  $p_{n-1}$  и т. д. Переменная  $u$  накапливает сумму  $b_n p_n + b_{n-1} p_{n-1} + \dots + b_1 p_1$ .

Определим изменение численности популяции за 10 лет:

```

> Таблица:= seq(T [i] ,i=1..m) :
  seq(round(T [i] ) ,i=1..m) ;

```


182, 207, 213, 225, 270, 331, 391, 453, 533, 631

Графическая иллюстрация решения:

```

> with(plots) :
gr1 := listplot([Таблица] , style=line, color=red, thickness=2) :
gr2 := listplot([Таблица] , style=point, color=blue,
  symbol=circle, symbolsize=18) :
display([gr1, gr2], view=[0..11, 150..650], axes=boxed,
  axis=[gridlines=[10, color=green]]);

```



Очевидно, что численность популяции быстро растет.

### Модель народонаселения

Модель Лесли можно использовать для прогнозирования изменений возрастной структуры населения.

По данным за некоторый год подсчитаем количество людей в следующих возрастных диапазонах:  $D_1$  — 0—5 лет;  $D_2$  — 5—15 лет;  $D_3$  — 15—30 лет;  $D_4$  — 30—60 лет;  $D_5$  — 60—70 лет;  $D_6$  — 70—80 лет. Людей старше 80 лет учитывать не будем. Заметим, что возрастные диапазоны могут быть и другими.

Сделаем некоторые предположения, упрощающие модель. Будем считать, что внутри одного возрастного диапазона в каждой годовой группе одинаковое количество людей. Например, диапазон 0—5 лет включает пять таких групп: 0—1, 1—2, 2—3, 3—4, 4—5 лет. Пусть в диапазоне  $D_i$  коэффициент рождаемости равен  $b_i$ , а смертности —  $s_i$ . Можем также предположить, что  $b_1 = b_2 = b_5 = b_6 = 0$ .

Пусть в диапазоне  $D_i$  ( $i \geq 2$ )  $k_i$  годовых групп и первоначально  $n_i$  человек, т. е. по  $n_i/k_i$  человек в группе. За год в диапазоне  $D_i$  умрет  $s_i n_i$  человек, т. е. по  $s_i n_i/k_i$  в каждой годовой группе. Из оставшихся  $n_i - s_i n_i = n_i(1 - s_i)$  человек в следующую годовую группу перейдут  $n_i(1 - s_i)/k_i$ . Из предыдущего диапазона в диапазон  $D_i$  перейдут  $n_{i-1}(1 - s_{i-1})/k_{i-1}$  человек. Тогда в диапазоне  $D_i$  через год будет

$$n_i - n_i(1 - s_i)/k_i + n_{i-1}(1 - s_{i-1})/k_{i-1} \text{ человек.}$$

Кроме того, в диапазоне  $D_i$  за год родится  $b_i n_i$  человек, которые окажутся в диапазоне  $D_1$ .

Вся эта информация представлена в табл. 1

Таблица 1

Возрастной диапазон	0—5	5—15	15—30
Число годовых групп в диапазоне	5	10	15
Количество людей в диапазоне	$n_1$	$n_2$	$n_3$
Умерли	$s_1 n_1$	$s_2 n_2$	$s_3 n_3$
Пережили	$n_1(1 - s_1)$	$n_2(1 - s_2)$	$n_3(1 - s_3)$
Перешли в другой диапазон	$n_1(1 - s_1)/5$	$n_2(1 - s_2)/10$	$n_3(1 - s_3)/15$
Пришли из других диа-	$b_3 n_3 + b_4 n_4$	$n_1(1 - s_1)/5$	$n_2(1 - s_2)/10$

пазонов			
Родилось	0	0	$b_3n_3$
Возрастной диапазон	30—60	60—70	70—80
Число годовых групп в диапазоне	30	10	10
Количество людей в диапазоне	$n_4$	$n_5$	$n_6$
Умерли	$s_4n_4$	$s_5n_5$	$s_6n_6$
Пережили	$n_4(1 - s_4)$	$n_5(1 - s_5)$	$n_6(1 - s_6)$
Перешли в другой диапазон	$n_4(1 - s_4)/30$	$n_5(1 - s_5)/10$	$n_6(1 - s_6)/10$
Пришли из других диапазонов	$n_3(1 - s_3)/15$	$n_4(1 - s_4)/30$	$n_5(1 - s_5)/10$
Родилось	$b_4n_4$	0	0

Количество людей через год в каждом из возрастных диапазонов нетрудно определить по следующей формуле:

Количество людей в диапазоне через год =  
= Количество людей диапазона, переживших этот год, –  
– Количество людей, покинувших диапазон, +  
+ Количество людей, перешедших из других диапазонов

Тогда через год количество людей в диапазонах определяется по формулам, приведенным в табл. 2.

**Таблица 2**

Диапазон	Формула
$D_1$	$n_1(1 - s_1) - n_1(1 - s_1)/5 + b_3n_3 + b_4n_4$
$D_2$	$n_2(1 - s_2) - n_2(1 - s_2)/10 + n_1(1 - s_1)/5$
$D_3$	$n_3(1 - s_3) - n_3(1 - s_3)/15 + n_2(1 - s_2)/10$
$D_4$	$n_4(1 - s_4) - n_4(1 - s_4)/30 + n_3(1 - s_3)/15$
$D_5$	$n_5(1 - s_5) - n_5(1 - s_5)/10 + n_4(1 - s_4)/30$
$D_6$	$n_6(1 - s_6) - n_6(1 - s_6)/10 + n_5(1 - s_5)/10$

Перейдем к реализации этой модели в Maple.

Пусть имеем такие исходные данные (количество людей в возрастных диапазонах, в млн чел.):

```

[> restart:
[> N0:= [0.5,0.9,2.3,4.2,0.9,0.7]:
      V:= [0.0,0.0,0.0011,0.007,0.0,0.0]:
      S:= [0.0013,0.0002,0.0013,0.009,0.032,0.0775]:
[> Начальное_распределение_населения = evalf(seq(N0[i],i=1..6),4);
      Начальное_распределение_населения = (0.5, 0.9, 2.3, 4.2, 0.9, 0.7)

```

Чтобы на основе нашей модели предсказать распределение населения по возрастным группам через  $t = 10$  лет, нужно повторить все вычисления 10 раз:

```

[> m:= 10: T:= array(0..m): T[0]:= sum(N0[i],i=1..6):
      Начальная_численность_населения = evalf(T[0],3);
      Начальная_численность_населения = 9.5
[> N:= N0:
      for i to m do
          n:= N:
          N[1]:= n[1]*(1-s[1])-n[1]*(1-s[1])/5+v[3]*n[3]+v[4]*n[4]:
          N[2]:= n[2]*(1-s[2])-n[2]*(1-s[2])/10+n[1]*(1-s[1])/5:
          N[3]:= n[3]*(1-s[3])-n[3]*(1-s[3])/15+n[2]*(1-s[2])/10:
          N[4]:= n[4]*(1-s[4])-n[4]*(1-s[4])/30+n[3]*(1-s[3])/15:
          N[5]:= n[5]*(1-s[5])-n[5]*(1-s[5])/10+n[4]*(1-s[4])/30:
          N[6]:= n[6]*(1-s[6])-n[6]*(1-s[6])/10+n[5]*(1-s[5])/10:
          T[i]:= sum(N[j],j=1..6):
      end do:

```

Распределение населения по возрастным группам через 10 лет:

```

[> Распределение_населения = evalf(seq(N[i],i=1..6),4);
      Распределение_населения = (0.1871, 0.6793, 1.750, 3.851, 0.9986, 0.5806)

```

В итоге наблюдаем убыль населения во всех возрастных диапазонах кроме пятого:

```

[> Динамика = evalf(seq(N[i]-N0[i],i=1..6),3);
      Динамика = (-0.313, -0.221, -0.55, -0.35, 0.099, -0.119)

```

Ежегодное изменение численности населения в течение 10 лет:


```
> Tab1:= seq(T[i],i=0..m):
Численность_населения = evalf(Tab1[1..m+1],3);

Численность_населения = (9.5, 9.34, 9.19, 9.04, 8.90, 8.75, 8.61, 8.47, 8.33, 8.19, 8.05)
```

Покажем эти значения на графике:

```
> with(plots):
data:= seq([i,T[i]],i=0..m):
gr1:= listplot([data],style=line,color=red,thickness=2):
gr2:= listplot([data],style=point,color=blue,
symbol=box,ybolsize=18):

> display([gr1,gr2],view=[-0.5..m+0.5,7.8..9.8],
axes=boxed,axis=[gridlines=[15,color=green]],
tickmarks=[[seq(0..10)],[seq(8..9.6,0.2)]]);
```



Из рисунка заключаем, что численность населения сокращается.

### Модель эпидемии

В небольшом городке начинается эпидемия гриппа. Санитарные службы заинтересованы в получении прогноза ее развития. Это позволит спланировать их работу так, чтобы быстрее победить болезнь. Можно подвезти лекарства в аптеки, отозвать врачей из отпуска, выписать из больниц выздоравливающих. Попробуем помочь им, моделируя развитие начавшейся эпидемии.



Пусть в городке живет  $N$  человек. В начальный период эпидемии инфицировано  $b_0$  человек. Остальные  $z_0 = N - b_0$  человек здоровы. Будем считать, что все, перенесшие заболевание, приобретают иммунитет к нему и больше не болеют. Поэтому число заболевших в первый день пропорционально произведению числа больных на число еще не болевших:  $z_1 = pb_0 z_0$ , где  $p$  — коэффициент пропорциональности, зависящий от различных мер профилактики. Как он определяется? Скорее всего, подобные эпидемии уже были и раньше, поэтому есть данные о том, как влияют на распространение инфекции профилактические прививки, изоляция заболевших, использование марлевых повязок и другие мероприятия.

Обозначим через  $b_i$  — число человек, заболевших в течении  $i$ -го дня, через  $z_i$  — число человек еще не болевших к концу этого дня. Тогда математическую модель эпидемии можно описать двумя уравнениями:

$$\begin{aligned} b_i &= pb_{i-1}z_{i-1}, \\ z_i &= z_{i-1} - b_i. \end{aligned}$$

Спрашивается, как развивается эпидемия, т. е. как ежедневно изменяется число  $b$  больных и число  $z$  не болевших?

Пусть в городке с населением  $N = 20\,000$  человек первоначально заболели 50, а коэффициент  $p = 0,0001$ .

Определим исходные данные:

```
> n:= 13: p:= 0.0001: N:= 20000:
  B:= array(0..n): Z:= array(0..n):
  B[0]:= 50.0: Z[0]:= N-50:
```

Развитие эпидемии:


```
> for i from 1 to n do
  B[i]:= p*B[i-1]*Z[i-1]:
  Z[i]:= Z[i-1]-B[i]:
end do:

> Больные:= seq(trunc(B[i]), i=1..n):
  НеБолевшие:= seq(trunc(Z[i]), i=1..n):
  [Больные [1..n]] ; [НеБолевшие [1..n]] ;

          [99, 198, 389, 749, 1387, 2376, 3505, 3941, 2878, 1273, 401, 110, 29]
          [19850, 19652, 19263, 18513, 17125, 14749, 11243, 7302, 4424, 3150, 2749, 2639, 2610]
```

При расчете количества больных и неболевших округляли получающиеся числа до целых значений.

Построим графики изменения количества больных и неболевших по дням:



Нижняя кривая показывает число больных, а верхняя — оставшихся здоровыми. Как видно из рисунка, на тринадцатый день наблюдается спад эпидемии — только 29 заболевших:

```
> Больные [13], НеБолевшие [13];
                                     29, 2610
```

Критическая точка — восьмой день — 3941 заболевший:

```
> Больные [8], НеБолевшие [8];
                                     3941, 7302
```

Чтобы определить всех, перенесших заболевание, нужно просуммировать число больных по дням:

```
> trunc(sum(B[j], j=0..n));
                                     17389
```

Прогнозирование развития эпидемии позволит лучше подготовиться к ней и принять необходимые профилактические меры.

### Задачи для самостоятельного решения

1. Реализовать модель загрязнения реки для всех вредных веществ (см. таблицу примера 2). Показать все графики на одном рисунке.

2. Построить модель загрязнения реки, в которой коэффициент самоочищения будет уменьшаться с течением времени.

3. Реализовать биологическую модель ограниченного роста из примера 3. Выяснить, как изменяется масса бактерий по дням в течение месяца для заданных значений параметров  $a$ ,  $b$ ,  $m$  и начальной массы  $x_0$ .

Известно, что для  $a = 1$ ,  $b = 0,0001$ ,  $m = 2000$  г и  $x_0 = 12000$  г общая масса бактерий сначала довольно быстро убывает, но через две недели становится равной приблизительно 7236 г, а после этого не изменяется. Если  $x_0 = 17000$  г, то масса бактерий сначала будет быстро расти, но к концу месяца она опять станет равной 7236 г. Если же увеличить начальную массу до 18 000 г, то уже через два дня практически все бактерии погибнут.

4. Разработать модель для изучения популяции кошек, описываемой возрастными диапазонами: 0—1, 1—6, 6—10 и 10—15 лет. Предполагается, что:

- кошки старше 15 лет не учитываются;
- рождения происходят в диапазоне 1—6 ( $b$  % в год);
- уровень смертности котят для диапазона 0—1 равен  $d_1$  % в год;
- в диапазоне 1—6 выживают все;
- уровень смертности для диапазона 6—10 равен  $d_5$  % в год, а для диапазона 10—15 —  $d_{15}$  % в год.

Найти решение для случая:  $b = 15$ ,  $d_1 = 5$  %,  $d_5 = 2$  %,  $d_{15} = 30$  %. Возрастет или уменьшится численность всей популяции за 5 лет, если исходная численность возрастных диапазонов: 20, 20, 15, 10?

5. Построить модель эпидемии, учитывающую отсутствие иммунитета.

6. Построить модель эпидемии, учитывающую, что на распространение болезни влияют все заболевшие, а не только те, кто заболели в последний день.

## 6. ПРОГНОСТИЧЕСКИЕ МОДЕЛИ

### 6.1. Примеры прогностических моделей

Модели, на основе которых по результатам изучения функционирования системы или течения процесса делают определенные выводы, позволяющие предсказать их дальнейшее развитие, называются *прогностическими*.

**Пример 1.** На предприятии, выпускающем 2 000 единиц продукции за смену, началась масштабная реконструкция, проходящая без остановки производства. Через три квартала предприятие выпускало уже 4 000 единиц продукции за смену, а через пять кварталов — 7 000. Каким будет выпуск продукции спустя еще два квартала, если реконструкция будет проходить так же как и ранее?

Самое сложное в прогностических моделях — перенести закономерности протекания процесса на его последующее развитие. В данном случае нужно разобраться, что означает фраза «реконструкция будет проходить также как и ранее» и формализовать ее.

Выскажем следующее предположение. С началом реконструкции, первые три квартала предприятие еще не приспособилось к производству продукции в новых условиях и поэтому не могло работать в полную силу, а после адаптации есть основания предполагать, что внедрение новой техники будет способствовать увеличению выпуска продукции в том же темпе.

Пусть ось  $x$  координатной плоскости соответствует кварталам, а ось  $y$  — количеству выпускаемой продукции. Отметим точки  $A(0; 2)$ ,  $B(3; 4)$  и  $C(5; 7)$ , которые показывают объем продукции соответственно до реконструкции предприятия, через три и пять кварталов после ее начала. (Ординаты точек показывают величину выпускаемой продукции в тысячах единиц.) Соединим эти точки отрезками прямой.

Если продолжить отрезок  $BC$ , то предполагаемый объем продукции через семь кварталов (считая от начала реконструкции) 10 000 единиц. Это ордината точки  $D$  на продолжении прямой  $BC$  с абсциссой, равной 7 (рис. 9).




Рис. 9

Это нетрудно и вычислить. Запишем уравнение прямой, проходящей через точки *B* и *C*, в таком виде:

$$y = ax + b.$$

Подставив в него координаты данных точек, получим систему двух уравнений относительно неизвестных *a* и *b*. Решив ее, определим эти коэффициенты:

$$\left[ \begin{array}{l} \text{> solve}(\{4=3*a+b, 7=5*a+b\}, \{a, b\}); \\ \{a = \frac{3}{2}, b = -\frac{1}{2}\} \end{array} \right.$$

Следовательно, если  $x = 7$ , то прогнозируемая производительность предприятия  $y = 10\ 000$  единиц продукции за смену:

$$\left[ \begin{array}{l} \text{> a:=3/2; b:=-1/2;} \\ \text{y:=a*x+b; subs(x=7, y);} \\ 10 \end{array} \right.$$

Достоинством этой модели является ее простота. Конечно, создавая модель, мы используем далеко не всю информацию о работе предприятия, поэтому нет уверенности, что реализованная гипотеза дает надежный прогноз.

На рост объема выпускаемой продукции будет влиять не только внедрение новой техники, важна также адаптация работников предприятия к новым условиям производ-

ства. Постараемся подобрать кривую, график которой проходит через точки  $A$ ,  $B$  и  $C$ . Выдвинем предположение, что эта кривая — парабола. Запишем уравнение параболы:

$$y = ax^2 + bx + c.$$

Подставив в него координаты точек  $A$ ,  $B$  и  $C$ , получим систему трех уравнений относительно неизвестных  $a$ ,  $b$  и  $c$ . Решив ее, найдем коэффициенты в уравнении параболы:

```

> eqvns := 2=a*0^2+b*0+c, 4=a*3^2+b*3+c, 7=a*5^2+b*5+c:
solve({eqvns}, {a, b, c});

```

$$\left\{ a = \frac{1}{6}, b = \frac{1}{6}, c = 2 \right\}$$

Вычислим значение  $y$  при  $x = 7$ :

```

> a:=1/6: b:=1/6: c:=2:
y:= a*x^2+b*x+c: subs(x=7,y): evalf(%,4);

```

11.33

и получим прогнозируемый объем выпуска продукции через семь кварталов после начала реконструкции (11 330 единиц за смену).

Для наглядности отобразим это на графике (рис. 10).




Рис. 10

Естественно возникает вопрос, какой из двух прогнозов более точный. Ответ существенно зависит от того, насколько правильны выдвинутые нами предположения о положительном влиянии реконструкции. Кроме того, при построении данной модели мы не учитывали возможное падение спроса на выпускаемую продукцию и соответствующую

щее снижение ее выпуска, изменение стоимости оборудования и материалов и т. д. Очевидно, что прогноз имеет оценочный характер, и сбудется он ли можно будет узнать только через два квартала.

Уточнить модель можно следующим образом. Мы знаем количество товаров, выпускаемое предприятием за смену спустя четыре квартала после реконструкции, сравнив его с результатами, полученными с помощью первой и второй модели, выберем ту, которая дает значение более близкое к реальному.

## 6.2. Эмпирические формулы

Если две величины зависят одна от другой, то можно измерить значения одной из них при определенных значениях другой. Будем рассматривать данные, полученные в результате измерений или наблюдений, как набор значений некоторой функции. Можно попробовать подобрать алгебраическое выражение, которое с достаточной точностью описывает зависимость между рассматриваемыми величинами. Это алгебраическое выражения называют *эмпирической формулой*.

Таким образом, эмпирические формулы подбирают опытным путем. Они должны давать хорошее совпадение полученных с их помощью результатов с экспериментальными данными. Поясним это на примерах.

**Пример 2.** Найти эмпирическую зависимость времени  $t$  валки деревьев от их диаметра  $d$  по результатам измерений, указанным в таблице.

$d$ , см	24	26	28	30	32	34	36	38	40	42
$t$ , с	49	53	54	56	61	62	65	69	72	75

Пусть горизонтальная ось координат соответствуют диаметру дерева  $d$ , а вертикальная — времени  $t$ . Если табличные данные в виде точек нанести на график (рис. 11), то можно сделать вывод, что они довольно хорошо ложатся на прямую линию.




Рис. 11

Как подобрать формулу, которая выражает предполагаемую линейную зависимость  $t(d)$  в виде функции? Для этого существуют специальные математические методы, которые не изучают в школе. Поэтому мы поступим проще.

Произвольно разделим точки, отмеченные на графике, на две группы: к первой отнесем, например, точки с координатами по оси  $x$ : 24, 26, 28, 30 и 32, ко второй — с координатами 34, 36, 38, 40 и 42. Затем определим такую точку  $A$ , координаты которой являются средним арифметическим координат точек первой группы, и такую точку  $B$ , координаты которой являются средним арифметическим координат точек второй группы.

Произведем вычисления для точки  $A$ :

```
> restart:
> dA:= (24+26+28+30+32)/5 : tA:= (49+53+54+56+61)/5:
  evalf( [dA, tA] , 4);
```

[28., 54.60]

и для точки  $B$ :

```
> dB:= (34+36+38+40+42)/5 : tB:= (62+65+69+72+75)/5:
  evalf( [dB, tB] , 4);
```

[38., 68.60]

После этого найдем коэффициенты уравнения прямой  $t = ad + b$ , проходящей через точки  $A$  и  $B$ , из системы уравнений:



```
> solve({tA=a*dA+b, tB=a*dB+b},{a,b}): evalf(%,3);
      {a=1.40, b=15.4}
```

Таким образом, получим эмпирическую формулу:

$$t = 1,4d + 15,4.$$

Воспользуемся этой формулой, чтобы узнать, за какое время можно повалить дерево диаметром 45 см:


```
> a:=1.4 : b:=15.4 : subs(d=45, a*d+b);
      78.4
```

Для наглядности начертим график, выражающую зависимости  $t(d)$ . Набор точек зададим списком:

```
> x:= 24,26,28,30,32,34,36,38,40,42:
   y:= 49,53,54,56,61,62,65,69,72,75:
   data:= seq([X[i],Y[i]],i=1..10):
```

и построим график:

```
> with(plots): with(plottools):
   gr1:= plot(a*d+b, d=23..43, color=blue):
   gr2:= plot([data], d=20..45, color=red,
             style=point, symbol=circle, symbolsize=16):
   display([gr1, gr2], view=[20..45, 40..80], axes=boxed,
           thickness=2, xtickmarks=[seq(20..45, 5)],
           axis=[gridlines=[10, color=green, linestyle=3]]);
```



Из рисунка видно, что хотя точки и расположены вблизи от полученной прямой, но не все лежат на ней. Это вызвано погрешностью измерения диаметра деревьев, различной плотностью древесины и другими факторами, которые трудно учесть.

Заметим также, что если по-другому распределить точки по группам, то при вычислении параметров  $a$  и  $b$ , которые входят в формулу, можно получить несколько иные значения.

**Пример 3.** При изучении химической реакции определили скорость  $v$  протекания реакции в зависимости от количества  $m$  одного из реагентов. Результаты измерений указаны в таблице и на графике (рис. 12).

<b><math>m</math></b>	5,0	5,6	6,0	6,4	6,8	7,2	7,6	8,0
<b><math>v</math></b>	9,2	14,4	17,1	22,3	25,9	32,1	38,4	46,0




Рис. 12

Определить формулу зависимости  $v$  от  $m$  и рассчитать скорость протекания реакции при  $m = 8,4$ .

Из рис. 13 видно, что экспериментальные точки не лежат на прямой, т. е. зависимость скорости химической реакции от количества реагента не является линейной функцией. Предположим, что эта зависимость отвечает степенной или показательной функции.

Выдвинем две гипотезы: зависимость  $v(m)$  в общем виде можно представить

$$v = ab^m \text{ или } v = am^b, \quad (6.1)$$

где  $a$ ,  $b$  — параметры, которые следует определить.

Логарифмируя первую из формул (6.1), получим

$$\ln v = \ln a + m \ln b.$$

Сделав замену  $y = \ln v$ ,  $c = \ln a$ ,  $d = \ln b$ , перейдем к новым переменным:

$$y = c + dm.$$

В результате имеем линейное уравнение относительно переменных  $m$  и  $y$ , а определять неизвестные коэффициенты линейной функции мы уже умеем.

Произведем все расчеты на рабочем листе Maple. Определим исходную таблицу с помощью последовательностей:


```
> restart:
> m:= 5.0,5.6,6.0,6.4,6.8,7.2,7.6,8.0:
> v:= 9.2,14.4,17.1,22.3,25.9,32.1,38.4,46.0:
```

Затем получим новую таблицу:

```
> y:= seq(ln(v[i]),i=1..8):
> evalf(m,3); evalf(y,3);
5.0, 5.6, 6.0, 6.4, 6.8, 7.2, 7.6, 8.0
2.22, 2.67, 2.86, 3.10, 3.25, 3.47, 3.65, 3.83
```

и отобразим табличные данные графически:

```
> with(plots): with(plottools):
> data:= seq([m[i],y[i]],i=1..8):
> gr1:= plot([data],x=5..8,color=blue,style=point,
> symbol=circle,symbolsize=16):
> display([gr1],view=[4.5..8.5,2..4],axes=boxed,
> thickness=2,axis=[gridlines=[10,color=green,linestyle=3]]);
```



Видим, что точки на рисунке расположены скорее по дуге, чем по прямой линии.

Таким образом, первая из гипотез не подтвердилась.


Прологарифмируем второе из выражений (6.1):

$$\ln v = \ln a + b \ln m.$$

После замены:  $y = \ln v$ ,  $c = \ln a$  и  $x = \ln m$ , получим линейную зависимость

$$y = c + bx. \quad (6.2)$$

Как и в первом случае, сформируем таблицу и построим график (рис. 14):



Видим, что на этот раз табличные точки достаточно хорошо выстроились вдоль некоторой прямой. Отсюда заключаем, что подтверждается вторая из предложенных гипотез.

Найдем коэффициенты линейной зависимости (6.2).

Как и прежде, выделим две группы точек: одну из них составят первые четыре, другую — оставшиеся. Определим точки  $A$  и  $B$ , координаты которых являются средними арифметическими координат точек первой и второй групп соответственно. Выполнив те же вычисления, найдем значения коэффициентов  $b$  и  $c$ :  $b = 3,36$ ;  $c = -3,17$ .

Из соотношения  $c = \ln a$  найдем  $a$ :

```
> a:=exp(-3.17): evalf(a,2);
```

0.042

Следовательно, эмпирическая формула в данном случае имеет вид:

$$v = 0,042m^{3,36}.$$

По ней легко определим скорость химической реакции при  $m = 8,4$ :

```
> v:=0.042*m^3.36: subs(m=8.4,v): evalf(%,4);
```

53.56

График на рис. 15 показывает, что все точки исходной таблицы данных хорошо ложатся на кривую, построенную по этим данным.




Рис. 13

Функции, которые используются для получения эмпирических формул, могут быть самыми разными. Надо только найти подходящую замену переменных, чтобы получить линейную зависимость. Примеры таких функций даны в табл. 3.

Таблица 3

Функция $y(x)$	Замена переменных	Линейная функция
$y = a/x + b$	$v = 1/x$	$y = av + b$
$y = a \ln x + b$	$v = \ln x$	$y = av + b$
$y = ae^{bx}$	$u = \ln y, c = \ln a$	$u = bx + c$
$y = ax^b$	$u = \ln y, v = \ln x, c = \ln a$	$u = bv + c$
$y = axe^{-bx}$	$u = \ln (y/x), c = -b, d = \ln a$	$u = cx + d$

### 6.3. Прогнозирование цены выпускаемого товара

**Пример 4.** Фирма приступает к выпуску новой продукции, и ее администрация решает провести исследование, чтобы определить оптимальное количество товара. С одной стороны, очевидно, что чем ниже цена, тем скорее будет продан товар. С другой стороны, чем выше будет цена, тем большую прибыль получит фирма и скорее окупит свои затраты. Однако слишком повышать цену нельзя: покупатель может отказаться от товара, и фирма потерпит убытки.

Исследования показали, что фирма может продать  $n(x) = 1600 - \frac{x^2}{6}$  единиц товара в квартал, где  $x$  — предполагаемая цена единицы товара, которая не может быть меньше 40 и больше 100 руб.

Как получена эта формула? Среди потенциальных покупателей провели опрос, чтобы выяснить, по какой наибольшей цене они готовы покупать товар. Была установлена табличная зависимость между предполагаемой ценой и возможной величиной продаваемого по этой цене товара. После этого подобрали эмпирическую формулу (см. 6.2).

Кроме этого, установили, что выпуск  $m(x) = 700 + 5x$  единиц товара в квартал принесет фирме необходимую прибыль, и исследовали разность между затратами на производство товара и выручкой от его продажи.

На (рис. 14) изображены графики функций  $n(x)$  и  $m(x)$ .

Точка  $A$  пересечения графиков называется *точкой равновесия*. Она определяет равновесие между ценой, устраивающей покупателей, и ценой, устанавливаемой производителем товара. Решив уравнение  $n(x) = m(x)$ , найдем значение  $x$  в этой точке, т. е. оптимальную цену товара:

```

> solve(1600-x^2/6=700+5*x, x) ;
-90, 60

```

Отрицательное значение не может быть решением поставленной задачи. Поэтому  $x = 60$ , тогда ордината точки  $A$ , т. е. количество выпускаемого товара:  $700 + 5 \cdot 60 = 1000$ .




Рис. 14

В этом случае выручка составит  $60 \cdot 1000 = 60\,000$  руб. Повышать цену единицы товара опасно, поскольку тогда фирма рискует не продать часть своей продукции.

Обратите внимание, что на количество выпускаемой продукции и на ее цену существенно влияют результаты исследования. Производители товара должны очень серьезно относиться к его проведению. Им предстоит ответить на непростой вопрос, кто может быть потенциальным покупателем продукции. Кроме того, важной характеристикой является число опрошенных клиентов. Чем оно больше, тем точнее результат исследования, однако при этом увеличиваются сроки его проведения и стоимость.

Отметим, что при построении модели не учитывались возможные колебания спроса на товар, действия конкурентов, появление на рынке аналогичного товара лучшего качества и целый ряд других факторов.

### Задачи для самостоятельного решения

1. В работах Д. И. Менделеева приведены данные о растворимости нитрата натрия  $\text{NaNO}_3$  в зависимости от температуры  $t$  воды. Измерения фиксируют, сколько условных частей (усл. ч.) у соли растворяется в 100 усл. ч. воды. Результаты содержатся в таблице.

$t, ^\circ\text{C}$	0	4	10	15	21	29	36	51	68
$y, \text{ усл. ч.}$	66,7	71,0	76,3	80,6	85,7	92,9	99,4	111,6	125,1

Найти зависимость  $y$  от  $t$ . Предполагается, что эта зависимость линейная.

2. Как известно, время  $t$  выполнения на компьютере «пузырьковой» сортировки числового набора данных быстро растет при увеличении количества  $n$  чисел. Предполагается, что зависимость  $t$  от  $n$  выражается формулой  $t(n) = Cn^2$ . Определить коэффициент  $C$  по результатам измерений, которые приведены в таблице.

$t, \text{ с}$	2,41	9,83	39,55
$n$	5 000	10 000	20 000

Сколько времени потребуется, чтобы отсортировать 100 000 чисел?

3. Для определения ускорения свободного падения  $g$  получены данные, приведенные в таблице.

$t, \text{ с}$	0,2	0,4	0,6	0,8	1,0
$s, \text{ м}$	0,1960	0,7850	1,7665	3,1405	4,9075

Найти значение  $g$  из соотношения  $s = gt^2/2$ .

4. Во время лечебного голодания масса тела  $P$  пациента за 30 дней снизился со 106 до 72 кг. Контрольные измерения дали результаты, указанные в таблице.

Дни	1	5	8	10	12	15	20	25	30
$P, \text{ кг}$	106,0	101,8	98,4	95,5	93,2	89,8	83,6	77,8	72,0

Представить их в виде эмпирической формулы и графически.



## 7. МОДЕЛИРОВАНИЕ СЛУЧАЙНЫХ ПРОЦЕССОВ

### 7.1. Случайные числа. Получение случайных чисел

**Пример 1.** На клетчатой бумаге, сторона одной клетки принимается равной единице длины, нарисованы два пересекающихся круга  $(x - 2)^2 + y^2 \leq 7$  и  $(x + 1)^2 + y^2 \leq 3$  (рис. 15). Найти площадь получившейся объединенной криволинейной фигуры.




Рис. 15

Существует достаточно простой способ приближенного вычисления площади такой фигуры. Подсчитывается количество клеток, полностью оказавшихся внутри фигуры, и количество клеток, которые пересекаются линиями ее контура. Первое число складывают со вторым, деленным пополам. В данном случае площадь фигуры будет равна  $16 + 24/2 = 28$  кв. ед. Если мы хотим получить более точный ответ, то надо взять клетчатую бумагу, на которой сторона клетки будет равна меньшей единице длины.

Рассмотрим еще одно решение. Поместим криволинейную фигуру в прямоугольник, площадь которого известна и равна  $P$ . Равномерно заполним прямоугольник  $n$  точками (рис. 16).




Рис. 16

Пусть  $m$  из них попадут внутрь фигуры. Можно считать, что отношение площади  $S$  фигуры к площади  $P$  содержащего ее прямоугольника равно отношению  $m$  к  $n$ . Отсюда получим приближенную формулу

$$S = \frac{m}{n} P.$$

Чем больше точек мы возьмем, тем точнее будет вычислена площадь фигуры. Если помещать точки на равном расстоянии друг от друга, то никакой принципиальной разницы в определении площади двумя рассмотренными способами нет. Например, можно разместить точки в серединах единичных квадратов (как это сделано на рис. 16). Тогда площадь фигуры, образованной пересекающимися кругами, будет  $(32/48) \cdot 48 = 32$  кв. ед.

Попробуем располагать точки в пределах выбранного прямоугольника случайно. Но как это сделать? Предположим, что прямоугольник является мишенью, в которую стреляют, целясь в ее центр. Конечно, пули будут попадать в случайные точки прямоугольника, но воспользоваться предложенным способом вычисления площади невозможно, поскольку пробоины, скорее всего, будут концентрироваться в середине мишени.

Следовательно, случайные точки должны быть еще и *равномерно распределены*. Это означает, что если разбить прямоугольник на любые прямоугольники равных площадей, то чем больше точек мы поместим в основной прямоугольник, тем отношение количества точек, попавших в любых два меньших прямоугольника, будет ближе к единице.

Аналогично понимается и равномерное распределение чисел на отрезке. Если разбить отрезок на любые отрезки равной длины, то чем больше точек будет на всем отрезке, тем отношение количества чисел, попавших в любых два маленьких отрезка, будет ближе к единице. Числа, распределенные подобным образом, называют *равномерно распределенными случайными* или *просто случайными*.

### Получение случайных чисел

Случайные числа можно получить разными способами. Например, поместим в мешочек 10 одинаковых шаров, на которых изображены цифры от 0 до 9. Перемешав шары, будем извлекать их по одному, каждый раз возвращая вынутый шар в мешочек и снова перемешивая. Цифры, имеющиеся на шарах, будем записывать. Четыре взятые подряд шара дают случайное четырехзначное число, следующие четыре — еще одно число и т. д. Таким образом можно получить таблицу четырехзначных случайных чисел.

Рассмотрим способ получения случайных чисел, меньших единицы. Перетасовав колоду из 52 карт, возьмем три верхние и вычислим значение выражения

$$\{[(x-1)/13 + y-1]/13 + z-1\}/13, \quad (7.1)$$

где  $x, y, z$  — числовые значения выбранных карт; считаем, что туз — 1, валет — 11, дама — 12, король — 13.

Например, если три верхних карты: семерка бубен, десятка червей и шестерка бубен, то  $x = 7, y = 10, z = 6$ , и в результате по формуле (7.1) получим случайное число 0,440601.

Конечно, при решении задач с помощью компьютера механические устройства получения случайных чисел заменяются программными «генераторами» или *датчиками* таких чисел. Это может вызвать недоумение: ведь все, что делает вычислительная машина каким-то образом заранее определено. Откуда же может появиться случайность?

*Моделировать* последовательности случайных чисел мы будем с помощью алгоритмов и формул. Ясно, что такие числа не могут быть случайными, но выглядят как случайные. Если при этом они удовлетворяют необходимому равномерному распределению, то их можно считать случайными.

Способ моделирования случайных чисел первым предложил Джон фон Нейман (1903—1957) — американский математик, один из основоположников кибернетики. Этот способ назвали *методом середины квадрата*. Вот один из его вариантов.

Берем произвольное начальное четырехзначное число, возводим его в квадрат и, если необходимо, дописываем слева нули, чтобы получить восьмизначное число. В качестве нового случайного числа берем средние четыре цифры этого восьмизначного числа и т. д.

Например, если начать с числа 1234, то получается такая последовательность чисел: 5227, 3215, 3362, 3030, 1809, 2724, 4201, 6484, ... . Как видно из примера, метод середины квадрата может довольно хорошо «перемешивать» числа. Однако он имеет существенный недостаток: если какое-либо число в последовательности окажется равным нулю, то и все последующие также будут нулями.

Рассмотрим еще один способ получения случайных чисел. Возьмем произвольное число  $x_0$  между 0 и 1. Последующие числа будем получать по формуле

$$x_{i+1} = \text{дробная часть } (11x_i + \pi), i = 0, 1, \dots \quad (7.2)$$

Так как берем дробную часть, то каждое новое число лежит между 0 и 1. Пусть, например,  $x_0 = 0,52$ . Тогда последовательность чисел, полученных по формуле (7.2), будет такой: 0,86159, 0,61911, 0,95182, ... . Этот метод может использоваться для формирования до 9000 неповторяющихся чисел, распределенных на отрезке  $[0, 1]$ .

Для получения чисел с равномерным распределением на произвольном отрезке  $[a, b]$  нужно выполнить дополнительные действия

$$y_i = (b - a)x_i + a,$$

где  $x_i$  — случайные числа с равномерным распределением на отрезке  $[0, 1]$ .

## 7.2. Гистограмма распределения случайных чисел

Может возникнуть вопрос: что получится, если в формуле (7.2) заменить коэффициенты или выбрать другое начальное число? Оказывается, что при некоторых исходных значениях последовательности могут оказаться «плохими».

Достаточно простым методом проверки равномерности распределения случайных чисел является гистограмма, которая отображает частоту попадания чисел в заданные интервалы. Построим гистограмму (см. 2.3). Разобьем отрезок  $[0, 1]$  на  $n$  равных интер-

валов (например,  $n = 20$ ). Вычислим большое количество  $m$  (например,  $m = 6000$ ) случайных чисел. Для каждого интервала на графике построим столбик, высота которого равна количеству чисел, попавших в этот интервал. Если случайные числа распределены равномерно, то все столбики гистограммы будут примерно одинаковой высоты.

Составим программу для проверки этого утверждения.

Для хранения частот используем массив счетчиков, первоначально все они равны нулю:

```
[> restart:
> n:= 20 : H:= array(1..n, [seq(0,i=1..n)]):

> x:= 0.52:
  for i from 1 to 6000 do
    x:= evalf(frac(11.0*x+Pi)):
    k:= trunc(x*n)+1:
    H[k] := H[k]+1
  end do:
```

Здесь  $\text{frac}(z)$  — дробная часть  $z$  (целая часть отбрасывается),  $\text{trunc}(z)$  — целая часть  $z$  (отбрасывается дробная часть). Номер интервала, в который попадает значение  $x$ , вычисляется по формуле

$$k = \text{trunc}(xn) + 1.$$

Например, если отрезок  $[0, 1]$  разделить на  $n = 10$  интервалов, то число  $x = 0,2$  попадает в шестой интервал: целая часть произведения  $0,52 \cdot 10$  равна 5.

Вычисленные частоты:

```
[> seq1:= seq(H[i], i=1..n):
  seq1[1..10]; seq1[11..20];


          306, 257, 297, 299, 277, 305, 306, 317, 324, 309
          286, 321, 298, 285, 307, 301, 274, 293, 314, 324
```

Ожидаемое среднее значение равно 300. Проверим это:

```
[> (sum(H[j], j=1..n))/n;

          300
```

Графическая иллюстрация в виде гистограммы:



Из рисунка видно, что столбики гистограммы несколько отличаются по высоте. Будет ли такое распределение равномерным? Это зависит от решаемой задачи. Для многих приложений его можно считать равномерным.

Из большого количества известных методов получения целых случайных чисел одним из лучших считается следующий алгоритм. Берут некоторое начальное целое число  $x_0$ . Последующие числа  $x_1, x_2, \dots$  получают по формуле

$$x_{i+1} = (ax_i + c) \bmod m, \quad 1 \leq a \leq m - 1, \quad i = 0, 1, \dots \quad (7.3)$$

Операция  $p \bmod q$  означает остаток от деления  $p$  на  $q$ . В формуле (7.3) обычно  $m$  — простое число,  $a$  — целое число,  $c$  — целое число или 0. Так как мы берем остатки от деления на  $m$ , то все получаемые числа лежат между 0 и  $m - 1$ . Дополнительное деление на  $m$  дает числа  $y_i = x_i/m$ , меньшие 1.

Для разных значений параметров  $a, c$  и  $m$  получают последовательности со своими особенностями. Например, для  $x_0 = a = c = 7$  и  $m = 10$  последовательность будет такой: 7, 6, 9, 0, 7, 6, 9, 0 и т. д. Отсюда видно, что неудачный выбор параметров в формуле (7.3) может привести к неудачной числовой последовательности.

Определены значения параметров  $m$  и  $a$ , которые дают хороший результат. Некоторые из них приведены в табл. 4.

Таблица 4

$m$	$a$
300 023	5682, 6525, 6754, 6790, 7000
1 000 667	1580, 1894, 2118
1 000 919	1373, 2076, 2084
1 001 003	1430, 1856, 2000, 2063

Проверим равномерность распределения 6000 случайных чисел на отрезке  $[0, 1]$ , если  $m = 300023$ ,  $a = 5682$ ,  $c = 0$ ,  $x_0 = 152000$ . Гистограмма распределения чисел по 20 интервалам дана на рис. 17.




Рис. 17

В системе Maple для получения целых равномерно распределенных случайных чисел предназначена функция `rand()`, с помощью которой можно получать числа, содержащие до 12 цифр, например:

```

> restart:
  rand(),rand(),rand();

395718860534, 193139816415, 22424170465

```

Если нужны числа, которые лежат между 0 и 1, то используют деление на  $10^{12}$ .

```

[ > restart:
  r:= rand(): evalf(r/10^12,5);

                                0.39572

```

Чтобы получить целые случайные числа в диапазоне от  $a$  до  $b$ , используется вспомогательная функция  $\mathbf{rnd} = \mathbf{rand}(a \dots b)$ . Пример формирования последовательности из 10 чисел, лежащих между 0 и 999:

```

[ > restart:
  rnd:= rand(0..999):
  seq(rnd(),i=1..10);

                                860, 758, 750, 889, 300, 991, 5, 993, 954, 299

```

### 7.3. Метод Монте-Карло

Метод решения математических задач путем моделирования и использования случайных чисел называется *методом Монте-Карло*. Название происходит от города Монте-Карло в княжестве Монако, знаменитого своими игорными домами. Наверно самой известной азартной игрой является рулетка. Рулетка — один из простейших механических приборов для получения случайных чисел. Метод Монте-Карло называют также *методом статистических испытаний*.

Теперь, когда у нас есть компьютерный генератор случайных чисел (функция  $\mathbf{rand}$ ), мы можем решать задачи методом Монте-Карло.

**Пример 2.** Рассмотрим способ использования случайных чисел для приближенного вычисления площади криволинейной фигуры, ограниченной дугой синусоиды и осью  $x$  на отрезке от 0 до  $\pi$ .

Прежде всего, определим прямоугольную область, которая заключает дугу синусоиды:

```

[ > restart:
  a:= 0: b:= Pi: c:= 1:

```

Площадь прямоугольной области равна  $\pi$ :

```

[ > P:= (b-a)*c: P;

                                 $\pi$ 

```



Формируем две последовательности, каждая из которых содержит, например, 6000 случайных чисел  $a \leq x_i \leq b$  и  $0 \leq y_i \leq 1$ :

```
> n:= 6000:
  X:= evalf(seq((b-a)*rand()/10^12+a,i=1..n)):
  Y:= evalf(seq(rand()/10^12,i=1..n)):
```

Подсчитаем количество  $m$  случайных точек  $(x_i, y_i)$ , оказавшихся под дугой синусоиды:

```
> m:= 0:
  for i from 1 to n do
    if Y[i]<=sin(X[i]) then m:= m+1 end if:
  end do:
```


```
> m, n;
                                     3824, 6000
```

Следовательно, площадь криволинейной фигуры приблизительно равна:

```
> evalf(P*m/n, 4);
                                     2.002
```

Для наглядности вычертим прямоугольную область с 2000 случайных точек и синусоиду:

```
> with(plots): with(plottools):
  gr1:= line([a,c],[b,c],linestyle=1,color=red):
  gr2:= line([b,0],[b,c],linestyle=1,color=red):
  gr3:= plot(sin(x),x=0..Pi,thickness=2,labels=["",""]):
  seq1:= seq(point([X[i],Y[i]],symbol=point),i=1..2000):
  > display([gr1,gr2,gr3,seq1],view=[0..3.2,0..1.05],color=blue,
    tickmarks=[spacing(Pi/2),[seq(0..1,1/2)]]);
```



**Пример 3.** С помощью метода Монте-Карло найти приближенное значение числа  $\pi$ .

Здесь воспользуемся способом, предложенным известным французским естествоиспытателем Жоржем Бюффоном (1707—1788) еще в середине XVIII века.

Начертим на листе бумаги параллельные прямые, находящиеся друг от друга на расстоянии 2. На лист будем бросать тонкую иглу такой же длины. Понятно, что каждый раз центр иглы окажется на некотором случайном расстоянии  $y$  от ближайшей к ней прямой, а сама игла образует случайный угол  $0 \leq \alpha < \pi$  с этой прямой (рис. 18).




Рис. 18

Конечно, существует бесконечно много положений иглы, определяемых величинами  $y$  и  $\alpha$  (рис. 19, а). Однако для всех таких положений игла или будет иметь общую точку с какой-либо прямой, или не будет ее иметь. Поэтому можно считать, что центр иглы всегда попадает на некоторый отрезок  $AB$  длиной 1, перпендикулярный параллельным прямым и расположенный между двумя из них, как показано на рис. 19, б.




Рис. 19

Таким образом, все возможные положения иглы определяются точками прямоугольника со сторонами  $\pi$  и 1 (рис. 20). Ордината  $0 \leq y \leq 1$  точки прямоугольника показывает расстояние центра иглы от прямой линии, а абсцисса  $0 \leq a \leq \pi$  — угол, который игла составляет с ней. Из рис. 19, б заключаем, что для пересечения иглы и прямой необходимо и достаточно выполнение условия  $y \leq \sin a$ . Это соответствует положению точки с координатами  $(a, y)$  в прямоугольнике ниже дуги синусоиды  $y = \sin a$ .




Рис. 20

Итак, можно не бросать иглу на разлинованную бумагу, достаточно заполнить случайными точками прямоугольник, который содержит полуволну синусоиды. Попадание случайной точки ниже кривой соответствует паданию иглы на прямую линию на листе бумаги.

Площадь  $P$  криволинейной фигуры, ограниченной синусоидой, была найдена методом Монте-Карлов примере 2;  $P \approx 2,002$ . Площадь прямоугольника равна  $\pi$ , количество  $n$  случайных точек — 6000, количество  $m$  точек, оказавшихся под синусоидой, — 3824. Из соотношения

$$\frac{\pi}{P} = \frac{n}{m}$$

получим приближенное значение числа  $\pi$ :

$$\pi = 2,002 \cdot \frac{6000}{3824} = 3,1412.$$

Как же поступал Бюффон, ведь у него не было компьютера для получения случайных чисел? Бюффон строил такую же модель, площадь фигуры определял аналитически с

помощью определенного интеграла (она равна 2), а затем бросал иглу. Если число произведенных бросков было равно  $n$ , а число попаданий на линию —  $m$ , то  $\pi$  определялось из соотношения

$$\pi = \frac{2n}{m}.$$

#### 7.4. Броуновское движение

В 1827 году английский ботаник Р. Броун (1773—1858) наблюдал удивительное явление, которое впоследствии назвали *броуновским движением*. Он увидел под микроскопом, что находящиеся в капле воды частицы цветочной пыльцы совершают беспорядочные движения. Сначала исследователь посчитал, что происходят они из-за каких-то внешних воздействий (колебаний стола, испарения воды и т. д.). Однако, несмотря на все старания, Броуну не удалось прекратить это движение, которое он подробно описал, но объяснить так и не смог. Вскоре было обнаружено, что так же ведут себя любые другие мелкие частицы, взвешенные в жидкости.

А объясняется броуновское движение следующим образом. Молекулы жидкости, которые находятся в непрерывном тепловом движении, при столкновении с частицей передают ей соответствующий импульс. Если размеры частицы достаточно малы, то толчки, получаемые частицей с разных сторон, не компенсируют друг друга. Из-за «перевеса» воздействия молекул с какой-либо одной стороны частица начинает двигаться.

Построим компьютерную модель броуновского движения малой частицы в сосуде с водой. На частицу со стороны окружающих ее молекул воды действует сила, случайно изменяющая свою величину и направление. Поэтому частица будет двигаться хаотично. Используя генератор случайных чисел, можно моделировать движение броуновской частицы.

Упрощенно представим сосуд, в котором находится частица, в виде прямоугольника. Начальное положение частицы — центр сосуда, который совпадает с началом координат. Пусть в результате перемещений частица оказалась в точке с координатами  $(x, y)$ . Определим случайные смещения частицы:  $0,5 < \Delta x < 0,5$  и  $-0,5 < \Delta y < 0,5$  вдоль координатных осей, и получим координаты  $(x + \Delta x, y + \Delta y)$  нового положения частицы. Если частица вышла за левую или правую границы прямоугольника, то вернем ее к бли-

жайшей внутренней точке, заменив смещение  $\Delta x$  на противоположное, т. е. на  $-\Delta x$ . То же самое, если частица вышла за нижнюю или верхнюю границы прямоугольника. Поместим частицу в точку с координатами  $(x + \Delta x, y + \Delta y)$  и т. д.

Реализацию алгоритма броуновского движения в Maple начнем с определения прямоугольной области, внутри которой будет двигаться частица:

```
[> restart:
[> left:=-3 : right:=3 : bottom:=-4 : top:=4:
```


Одна из возможных траекторий частицы после 400 столкновений вычисляется с помощью следующей программы (для хранения координат частицы используются массивы  $X$  и  $Y$ ):

```
[> n:= 400 : X:= array(0..n) : Y:= array(0..n):
x:= 0 : y:= 0:
X[0]:= x: Y[0]:= y:
for i from 1 to n do
  vx:= rand()/10^12-0.5: vy:= rand()/10^12-0.5:
  if x+vx>right then x:= right : vx:=-vx end if:
  if x+vx<left then x:= left : vx:=-vx end if:
  if y+vy>top then y:= top : vy:=-vy end if:
  if y+vy<bottom then y:= bottom : vy:=-vy end if:
  x:= x+vx : y:= y+vy:
  X[i]:= x : Y[i]:= y:
end do:
```

Вот как это будет выглядеть на графике:

```
[> data:= seq([X[i],Y[i]], i=0..n):
[> with(plots) :
with(plottools):
gr1:= plot([data],color=red):
gr2:= plot([data],color=blue,
style=point,symbol=circle,symbolsize=8):
gr3:= line([left,bottom],[left,top+1],color=blue,thickness=3):
gr4:= line([right,bottom],[right,top+1],color=blue,thickness=3):
gr5:= line([left,bottom],[right,bottom],color=blue,thickness=3):
gr6:= line([left,top],[right,top],color=blue,thickness=1):
```

```
> display([seq(gr||i,i=1..6)],view=[left..right,bottom..top+1],
tickmarks=[[ ],[ ]]);
```



### Задачи для самостоятельного решения

1. Провести эксперимент по получению случайных чисел с помощью колоды из 52 карт.
2. Построить гистограмму распределения случайных чисел, полученных по формуле (7.3) с параметрами, взятыми из табл. 4.
3. Вычислить с помощью метода Монте-Карло площадь пересечения двух кругов  $(x - 2)^2 + y^2 \leq 7$  и  $(x + 1)^2 + y^2 \leq 3$ . Решение проиллюстрировать графически.
4. Определить число  $\pi$  с помощью метода Монте-Карло, вычисляя площадь круга.
5. На горизонтальную плоскость, расчерченную параллельными прямыми, отстоящими друг от друга на расстоянии  $2a$ , случайно бросается круг, радиус которого меньше  $a$ . С помощью метода Монте-Карло определить частоту попадания круга на линию.
6. На горизонтальную плоскость, расчерченную параллельными прямыми, отстоящими друг от друга на расстоянии  $2a$ , случайно бросается квадрат, сторона которого меньше  $a$ . С помощью метода Монте-Карло определить частоту попадания квадрата на линию.

7. Построить модель броуновского движения частицы для трехмерного случая. Решение проиллюстрировать, построив график.
8. Построить модель броуновского движения частицы с учетом ее массы.

РЕПОЗИТОРИЙ БГПУ

## РЕКОМЕНДУЕМАЯ ЛИТЕРАТУРА

*Абрамов, С. А.* Начала информатики / С. А. Абрамов, Е. В. Зима. — М. : Наука, 1989. — 256 с.

*Арсак, Ж.* Программирование игр и головоломок / Ж. Арсак. — М. : Наука, 1990. — 224 с.

*Быкадоров Ю. А.* Информатика: учеб. пособие для 8—9 кл. общеобразоват. школ с рус. яз. обучения / Ю. А. Быкадоров, А. Т. Кузнецов. — Минск : Нар. асвета, 2000. — 544 с.

*Быкадоров Ю. А.* Информатика и вычислительная математика: учеб. пособие для 10—11 кл. общеобразоват. школ с повышенным уровнем изучения информатики с рус. яз. обучения / Ю. А. Быкадоров, А. Т. Кузнецов, А. Т. Щерба. — Минск : Нар. асвета, 1997. — 334 с.

*Васильев А. Н.* Maple 8 : самоучитель / А. Н. Васильев. — М. : Диалектика, 2003. — 352 с.

*Говорухин, В. Н.* Введение в Maple. Математический пакет для всех / В. Н. Говорухин. — М. : Мир, 1997. — 208 с.

*Доморяд, А. П.* Математические игры и развлечения / А. П. Доморяд. — М. : Гос. изд.-во физ.-мат. литературы, 1961. — 267 с.

*Совертков, П. И.* Занимательное компьютерное моделирование в элементарной математике: учеб. пособие / П. И. Совертков. — М. : Гелиос АРВ, 2004. — 384 с.